

COMPUTATIONS CONCERNING THE REGULARITY NUMBER OF ALMOST SIMPLE GROUPS

MARINA ANAGNOSTOPOULOU-MERKOURI AND TIMOTHY C. BURNES

ABSTRACT. In this supplementary file we present the MAGMA and GAP computations used in our main paper [1] on the regularity number of almost simple groups with socle an alternating or sporadic group. Throughout, we freely adopt the notation and terminology introduced in [1].

CONTENTS

1.	MAGMA functions	1
1.1.	Random search	1
1.2.	Regular orbits	2
1.3.	Soluble subgroups	4
2.	GAP functions	6
2.1.	Probabilistic methods	6
3.	Symmetric and alternating groups	9
3.1.	Regularity number	9
3.2.	Primitive subgroups	10
3.3.	Imprimitive subgroups	11
3.4.	Soluble subgroups	12
4.	Sporadic groups	13
4.1.	The regularity number	13
4.2.	Soluble triples	26
4.3.	Soluble maximal pairs	39
4.4.	Nilpotent-soluble pairs	41
	References	44

1. MAGMA FUNCTIONS

We begin by presenting several MAGMA [2] functions for studying the regularity of core-free tuples of subgroups of a finite group. We use version 2.28-4 of MAGMA for all of the computations.

1.1. Random search. The function `RandomReg` takes as input a group G , an ordered tuple $M = [H_1, \dots, H_t]$ of core-free subgroups of G , a positive integer N and a set A of k -tuples of integers of the form $[a_1, \dots, a_k]$, where $k \geq 2$ and $1 \leq a_i \leq t$ for all i .

The function outputs a subset B of A comprising the tuples $[a_1, \dots, a_k]$ in A for which random search (with at most N attempts) has failed to find elements $g_i \in G$ such that

$$\bigcap_{i=1}^k H_{a_i}^{g_i} = 1.$$

In other words, the output corresponds to a list of candidate non-regular k -tuples, where each component subgroup is in M .

```

RandomReg := function(G,M,N,A)
B:=[];
if #A ne 0 then
k:=#A[1];
for i in [1..#A] do
a:=A[i];
z:=0;
for j in [1..N] do
X:=Random(G) : i in [1..k-1];
L:=M[a[1]];
for l in [2..k] do
L:=L meet M[a[l]]^X[l-1];
end for;
if #L eq 1 then
z:=1;
break j;
end if;
end for;
if z eq 0 then
Append(~B,a);
end if;
end for;
end if;
return B;
end function;

```

1.2. Regular orbits. Our next function `RegOrbits` takes as input a group G and a k -tuple $S = [H_1, \dots, H_k]$ of core-free subgroups of G , where $k \geq 2$. It returns the number of regular orbits of H_k on $G/H_1 \times \dots \times G/H_{k-1}$. Note that S is regular if and only if H_k has at least one regular orbit.

```

RegOrbits := function(G,S)
f:=[];
for i in [1..#S-1] do
g,K,L:=CosetAction(G,S[i]);
Append(~f,g);
end for;
t:=1;
L:=[S[#S]];
while t le #S-1 do
L1:=[];
for i in [1..#L] do
R:=f[t](L[i]);
o:=Orbits(R);
L1:=[Stabilizer(R,Representative(o[j]))@f[t] : j in [1..#o]];
end for;
L:=L1;
t:=t+1;
end while;
return #[i : i in [1..#L] | #L[i] eq 1];
end function;

```

We now use `RegOrbits` in the following function `RegPlus`. Here the input is a group G , an ordered tuple $M = [H_1, \dots, H_t]$ of core-free subgroups of G and a set A of k -tuples of integers $[a_1, \dots, a_k]$, where $k \geq 2$ and $1 \leq a_i \leq t$ for all i . The output is the complete set of tuples $[a_1, \dots, a_k]$ in A such that $(H_{a_1}, \dots, H_{a_k})$ is non-regular.

```

RegPlus := function(G,M,A)
B:=[];
if #A ne 0 then
k:=#A[1];
for a in A do

```

```

S:=M[a[i]] : i in [1..k];
if RegOrbits(G,S) eq 0 then
  Append(~B,a);
end if;
end for;
end if;
return B;

end function;

```

We are in a position to present our main function for finding non-regular tuples, which combines all of the above functions. The input for `RegTuples` is a group G , an ordered tuple $M = [H_1, \dots, H_t]$ of core-free subgroups of G , a positive integer N and a set A of k -tuples of integers $[a_1, \dots, a_k]$, where $k \geq 2$ and $1 \leq a_i \leq t$ for all i . The output is the complete set of tuples $[a_1, \dots, a_k]$ in A such that $(H_{a_1}, \dots, H_{a_k})$ is non-regular.

To do this, the function first identifies any tuples $[a_1, \dots, a_k]$ in A such that

$$\prod_{i=1}^k |H_{a_i}| > |G|^{k-1},$$

which immediately implies that $(H_{a_1}, \dots, H_{a_k})$ is non-regular. We then use `RandomReg` to try to show that the remaining tuples are regular by finding random conjugates of the component subgroups with trivial intersection (up to at most N attempts per tuple). Finally, we apply the function `RegPlus` on the set of tuples for which random search has failed, and the complete list of non-regular tuples in A is returned as output.

```

RegTuples := function(G,M,N,A)

B:=[];
if #A ne 0 then
  k:=#A[1];
  A1:=[];
  for a in A do
    if &*[#M[i] : i in a] gt #G^(k-1) then
      Append(~B,a);
    else
      Append(~A1,a);
    end if;
  end for;

  D:=[];
  if #A1 ne 0 then
    C:=RandomReg(G,M,N,A1);
    if #C ne 0 then
      D:=RegPlus(G,M,C);
    end if;
  end if;

  if #D ge 1 then
    for d in D do
      Append(~B,d);
    end for;
  end if;

return B;
end function;

```

The following function `RegTuplesPlus` will also be useful. Here the input is a finite group G , an ordered tuple $M = [H_1, \dots, H_t]$ of subgroups of G and positive integers ℓ and N with $\ell \geq 2$. The output is the minimal integer $k \geq \ell$ such that every k -tuple of subgroups in M is regular, together with the number of non-regular $(k-1)$ -tuples with components in M (up to ordering) and the set of such $(k-1)$ -tuples.

```

RegTuplesPlus := function(G,M,l,N)

C:=[];
k:=1;
A1:=CartesianPower([1..#M],k);
A1:=[[x[i]: i in [1..k]] : x in A1];
A1:=[x : x in A1 | x eq Sort(x)];

R:=RegTuples(G,M,N,A1);
Append(~C,R);

while #R ne 0 do
  k:=k+1;
  A1:=[];
  for r in R do
    for i in [1..#M] do
      s:=Append(r,i);
      Append(~A1,s);
    end for;
  end for;
  A1:={Sort(x) : x in A1};
  A1:= [x : x in A1];
  R:=RegTuples(G,M,N,A1);
  if #R ne 0 then
    Append(~C,R);
  end if;
end while;

E:=[];
for d in C[#C] do
  F:=[];
  for i in d do
    Append(~F,M[i]);
  end for;
  Append(~E,F);
end for;

return k, #E, E, C[#C];
end function;

```

1.3. Soluble subgroups. We will also use some functions that are specifically designed to handle tuples of soluble subgroups. Recall that if G is a finite group with soluble radical $R(G) = 1$, then we define $R_{\text{sol}}(G)$ to be the minimal k such that every k -tuple of core-free soluble subgroups of G is regular.

Our first function `Sol` takes as input a finite group G with $R(G) = 1$ and a positive integer N . It returns the integer $R_{\text{sol}}(G) = k$. In addition, if $k \geq 3$ then it also returns the number of non-regular soluble $(k - 1)$ -tuples (up to conjugacy and ordering), as well as the actual tuples themselves (and if $k = 2$, it returns 0 and an empty tuple).

```

Sol := function(G,N)

M:=SolubleSubgroups(G);
M:=[M[i]'subgroup : i in [1..#M] | #PrimeFactors(#M[i]'subgroup) ge 2];
P:=PrimeFactors(#G);
for p in P do
  Append(~M,SylowSubgroup(G,p));
end for;

return RegTuplesPlus(G,M,2,N);
end function;

```

Notice that the previous function relies on MAGMA's ability to execute the command `SolubleSubgroups`, which is not going to be possible for several sporadic groups we need to consider in the proof of [1, Theorem 2]. In some of these more difficult cases, our next function `MaxSolubleSubgps` can be used as a substitute. Here the input is a finite group G ,

a subgroup H and a positive integer d . The output is a certain collection S of subgroups of H , denoted $\mathcal{S}_d(H)$ in [1, Definition 4.14], which has the property that if $K \leq H$ is soluble, then $K^g \leq L$ for some $g \in G$ and $L \in S$.

If H is soluble, then $S = \{H\}$, so let us assume H is insoluble. For each integer $1 \leq k \leq d$, let $\mathcal{C}_k(H)$ be the set of subgroups L of H with the property that there exists a chain of subgroups

$$L = L_0 < L_1 < L_2 < \cdots < L_{k-1} < L_k = H$$

where L_1 is insoluble and each L_i is a maximal subgroup of L_{i+1} . For instance, $\mathcal{C}_1(H)$ is the set of maximal subgroups of H . In addition, let $\mathcal{D}_k(H)$ be the set of soluble subgroups in $\mathcal{C}_k(H)$, and let $\mathcal{P}(H)$ be the set of all non-Sylow p -subgroups of H , ranging over all prime divisors p of $|H|$. Then S is defined to be a complete set of representatives of the G -classes of subgroups in the set

$$\left(\mathcal{C}_d(H) \cup \bigcup_{k=1}^{d-1} \mathcal{D}_k(H) \right) \setminus \mathcal{P}(H).$$

Here the key property is the fact that if $K \leq H$ is soluble, then $K^g \leq L$ for some $g \in G$ and $L \in S$.

```

MaxSolubleSubgps := function(G,H,d)

a:=1;
S:=[];
NS:=[];
if IsSoluble(H) then
  Append(~S,H);
else
  Append(~NS,H);
end if;

while (#NS ge 1) and (a le d) do
  NS1:=[];
  for H in NS do
    M1:=MaximalSubgroups(H);
    M1:=[M1[i]'subgroup : i in [1..#M1]];
    for i in [1..#M1] do
      b:=exists(K){K : K in S | IsConjugate(G,K,M1[i])};
      if IsSoluble(M1[i]) and (b eq false) then
        Append(~S,M1[i]);
      else
        Append(~NS1,M1[i]);
      end if;
    end for;
  end for;
  NS:=NS1;
  a:=a+1;
end while;

if #NS ge 1 then
  for H in NS do
    b:=exists(K){K : K in S | IsConjugate(G,K,H)};
    if b eq false then
      Append(~S,H);
    end if;
  end for;
end if;

S:=[S[i] : i in [1..#S] | #PrimeFactors(#S[i]) ge 2];
P:=PrimeFactors(#H);
for p in P do
  Append(~S,SylowSubgroup(H,p));
end for;

D:=[#S[i] : i in [1..#S]];
ParallelSort(~D, ~S);

```

```

S:=Reverse(S);
return S;

end function;

```

We also define the following function `SolMax`. The input is a finite group G and a positive integer N ; the function computes $R_{\text{sol max}}(G)$ (if G does not contain a soluble maximal subgroup, then $R_{\text{sol max}}(G)$ is not defined and the code returns “0”).

```

SolMax := function(G,N)

M:=MaximalSubgroups(G);
M:=[M[i]‘subgroup : i in [1..#M] | IsSoluble(M[i]‘subgroup) and #Core(G,M[i]‘subgroup) eq 1];

if #M eq 0 then
  a:=0;
  b:=0;
  c:=[];
else
  a,b,c:=RegTuplesPlus(G,M,2,N);
end if;

return a, b, c;
end function;

```

2. GAP FUNCTIONS

In this section we present our main GAP [6] functions, working with version 4.11.1. These functions rely on the GAP Character Table Library [3] in order to access stored character tables and associated fusion maps, and they play a key role in the proof of [1, Theorem 2] on sporadic groups.

2.1. Probabilistic methods. Let G be a finite group and let $\tau = (H_1, \dots, H_k)$ be a k -tuple of core-free subgroups of G . As in [1, Section 2.1], let $Q(G, \tau)$ be the probability that a randomly chosen element in $G/H_1 \times \dots \times G/H_k$ lies in a regular orbit of G and note that τ is regular if and only if $Q(G, \tau) < 1$. In [1, Lemma 2.1] we established the bound $Q(G, \tau) \leq \widehat{Q}(G, \tau)$, where

$$\widehat{Q}(G, \tau) = \sum_{i=1}^m |x_i^G| \cdot \left(\prod_{j=1}^k \text{fpr}(x_i, G/H_j) \right)$$

and x_1, \dots, x_m is a set of representatives of the conjugacy classes in G of elements of prime order.

Suppose the character table of G is available in [3] and the GAP function `Maxes` is executable, which allows us to access the character table of every maximal subgroup of G . In addition, let us assume that the function `FusionConjugacyClasses` is effective with respect to the character table of every maximal subgroup of G . Then we can use GAP to compute $\widehat{Q}(G, \tau)$ precisely for any given k -tuple τ of maximal subgroups of G .

Remark 2.1. Let G be an almost simple sporadic group. In every case, the character table of G is available in [3] and we can use `Maxes` to access the character table of every maximal subgroup, unless $G = \mathbb{M}$ is the Monster. In addition, if $G \neq \mathbb{B}, \mathbb{M}$, then the fusion map from H -classes to G -classes is available in [3] for every maximal subgroup H of G . The same is true when $G = \mathbb{B}$ is the Baby Monster, with the single exception of the case $H = (2^2 \times F_4(2)):2$, for which the fusion map is not known.

Our first function `QProb` takes as input the character table t of a finite group G , a tuple $M = [t_1, \dots, t_r]$ of character tables of subgroups of G (where t_i is the character table of H_i), the tuple $F = [f_1, \dots, f_r]$, where f_i is the fusion map from H_i -classes to G -classes, and a set

A of k -tuples of positive integers $[a_1, \dots, a_k]$, where $1 \leq a_i \leq r$ for all i . The function returns the complete set of k -tuples $[a_1, \dots, a_k]$ in A such that $\widehat{Q}(G, \tau) \geq 1$ for $\tau = (H_{a_1}, \dots, H_{a_k})$.

```

QProb := function(t,M,F,A)

local S, O, r, B, A2, A1, a, a1, P, i, S0, t1, D, z, G, j, S1, fus, c, k, b;

S:=SizesConjugacyClasses(t);;
O:=OrdersClassRepresentatives(t);;
r:=Size(A[1]);;

B:=[];
A2:=[];
for a in A do
  A1:=[];
  for i in a do
    Add(A1,Size(CharacterTable(M[i])));;
  od;
  a1:=Product(A1);;
  if a1 <= Size(t)^(r-1) then
    Add(A2,a);;
  else
    Add(B,a);;
  fi;
od;

if Size(A2) = 0 then
  B:=A;
else
  P:=[];
  for i in [1..Size(O)] do
    if IsPrime(O[i]) then
      Add(P,i);;
    fi;
  od;

  S0:=[];
  for i in [1..Size(M)] do
    t1:=CharacterTable(M[i]);;
    Add(S0,SizesConjugacyClasses(t1));;
  od;

  for D in A2 do
    z:=0;
    for i in P do
      a:=S[i];;
      G:=[];
      for j in D do
        S1:=S0[j];;
        fus:=F[j];;
        c:=0;
        for k in [1..Size(fus)] do
          if fus[k] = i then
            c:=c+S1[k];;
          fi;
        od;
        Add(G,c/a);;
      od;
      b:=Product(G);;
      if Size(G) > 0 then
        z:=z+a*b;
      fi;
    od;
    if z >= 1 then
      Add(B,D);;
    fi;
  od;
fi;

```

```
Sort(B);;
return B;
end;
```

We now use `QProb` in the function `QGAP` given below. As before, the function takes t , M and F as input, together with an integer $k \geq 2$. For all $\ell \in \{k, k+1, k+2\}$ it returns a complete list of the ℓ -tuples τ with component subgroups in M such that $\widehat{Q}(G, \tau) \geq 1$. See the example given below.

```
QGAP := function(t,M,F,k)
local B,C,i,P,A,p,q,B1,b,j,C1,c,D;

B:= [1..Size(M)];;
C:= [];;
for i in [1..k] do
  Add(C,B);
od;
P:=Cartesian(C);;
A:= [];;
for p in P do
  q:=ShallowCopy(p);;
  Sort(p);;
  if q = p then
    Add(A,p);
  fi;
od;

B:=QProb(t,M,F,A);;

if Size(B) > 0 then
  B1:= [];;
  for i in [1..Size(B)] do
    b:=B[i];;
    for j in [1..Size(M)] do
      Add(B1,Concatenation(b,[j]));;
    od;
  od;
  for p in B1 do
    Sort(p);
  od;
  B1:=Set(B1);;
  C:=QProb(t,M,F,B1);;
else
  C:= [];;
fi;

if Size(C) > 0 then
  C1:= [];;
  for i in [1..Size(C)] do
    c:=C[i];;
    for j in [1..Size(M)] do
      Add(C1,Concatenation(c,[j]));;
    od;
  od;
  for p in C1 do
    Sort(p);
  od;
  C1:=Set(C1);;
  D:=QProb(t,M,F,C1);;
else
  D:= [];;
fi;

Sort(B);;
Sort(C);;
```



```
Sort(D);;
return [B,C,D];
end;
```

Example 2.2. Suppose $G = M_{11}$. Then the following code return the complete set of maximal ℓ -tuples with $\widehat{Q}(G, \tau) \geq 1$ and $\ell \in \{4, 5, 6\}$.

```
t:=CharacterTable("M11");;
k:=4;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[1]; q[2]; q[3];
```

This returns

```
[ [ 1, 1, 1, 1 ], [ 1, 1, 1, 2 ], [ 1, 1, 2, 2 ], [ 1, 2, 2, 2 ],
  [ 1, 2, 2, 4 ], [ 2, 2, 2, 2 ], [ 2, 2, 2, 4 ] ]

[ [ 2, 2, 2, 2, 2 ] ]

[ ]
```

It follows that every maximal 6-tuple is regular and thus $R(G) \leq 6$. In addition, we see that $\tau = (H, H, H, H, H)$ is the only maximal 5-tuple with $\widehat{Q}(G, \tau) \geq 1$, up to conjugacy, where $H = L_2(11)$.

3. SYMMETRIC AND ALTERNATING GROUPS

3.1. Regularity number. Let G be an almost simple group with socle $T = A_n$ and recall that the *regularity number* of G , denoted $R(G)$, is the minimal integer k such that every core-free k -tuples of subgroups of G is regular.

Proposition 3.1. *Let $G = S_n$ or A_n with $5 \leq n \leq 12$. Then $R(G) = n - |S_n : G|$.*

Proof. The MAGMA code below outputs $[n, R(S_n), R(A_n)]$ for all n in the given range. To do this, we use the function `RegTuplesPlus` (see Section 1.2), setting $l = \lfloor n/2 \rfloor$ and taking M to be the set $\mathcal{M}'(G)$ of representatives of the G -classes of core-free maximal subgroups of G , together with representatives of any maximal subgroups H of A_n with $N_G(H) = H$ (see [1, Remark 2.10]).

```
for n in [5..12] do
  G:=Sym(n);
  T:=Socle(G);
  l:=n div 2;

  M:=MaximalSubgroups(G);
  M:=[M[i]'subgroup : i in [1..#M] | #Core(G,M[i]'subgroup) eq 1];
  M1:=MaximalSubgroups(T);
  M2:=[M1[i]'subgroup : i in [1..#M1]];
  M1:=[M1[i]'subgroup : i in [1..#M1] | #Normaliser(G,M1[i]'subgroup) eq #M1[i]'subgroup];

  for H in M1 do
    a:=exists(K){K : K in M | IsConjugate(G,H,K)};
    if a eq false then
      Append(~M,H);
    end if;
  end for;

  D:=[#M[i] : i in [1..#M]];
  ParallelSort(~D, ~M);
```

```

M:=Reverse(M);

a1,a2,a3,a4:=RegTuplesPlus(G,M,1,100);
b1,b2,b3,b4:=RegTuplesPlus(T,M2,1,100);
[n,a1,b1];
end for;

```

□

Proposition 3.2. *We have $R(G) = 4$ for every almost simple group $G \neq A_6, S_6$ with socle A_6 .*

Proof. We use the following code to compute the regularity number of the three relevant groups (namely, $\mathrm{PGL}_2(9)$, M_{10} and A_6 .²²).

```

G:=AutomorphismGroupSimpleGroup("A6");
T:=Socle(G);
M2:=MaximalSubgroups(T);
M2:=[M2[i]'subgroup : i in [1..#M2]];
S:=LowIndexSubgroups(G,#G div #T);
for j in [2,3,5] do
  G:=S[j];
  M1:=MaximalSubgroups(G);
  M1:=[M1[i]'subgroup : i in [1..#M1]];
  M1:=[H : H in M1 | #Core(G,H) eq 1];
  if j lt 5 then
    M3:=[H : H in M2 | #Normalizer(G,H) eq #H];
    M:=M1 cat M3;
  else
    M:=M1;
    for k in [2..4] do
      M3:=MaximalSubgroups(S[k]);
      M3:=[M3[i]'subgroup : i in [1..#M3] | #Core(G,M3[i]'subgroup) eq 1];
      M:=M cat M3;
    end for;
  end if;

  a1,a2,a3,a4:=RegTuplesPlus(G,M,2,100);
  [j,a1];
end for;

```

□

3.2. Primitive subgroups. Let G be an almost simple group with socle $T = A_n$ and recall that a subgroup H of G is *primitive* if $H \cap T$ acts primitively on $\{1, \dots, n\}$. We define $R_{\mathrm{prim}}(G)$ to be the minimal integer k such that every k -tuple of core-free primitive subgroups of G is regular (if G does not have a core-free primitive subgroup, then $R_{\mathrm{prim}}(G)$ is undefined).

Suppose $G \in \{S_n, A_n\}$ with $5 \leq n < 60$. The following MAGMA code returns

$$[n, R_{\mathrm{prim}}(S_n), R_{\mathrm{prim}}(A_n)]$$

for all n in the required range. In addition, $X[n] = [A, B]$, where A is the set of non-regular $(k-1)$ -tuples for S_n (with components in $\mathcal{M}'(G)$) where $k = R_{\mathrm{prim}}(S_n)$, and similarly B is the set of non-regular maximal $(\ell-1)$ -tuples for A_n , where $\ell = R_{\mathrm{prim}}(A_n)$.

```

X:=[[ ], [ ], [ ], [ ]];
for n in [5..59] do
  G:=Sym(n);
  T:=Alt(n);
  M1:=MaximalSubgroups(G);
  M2:=MaximalSubgroups(T);
  M1:=[M1[i]'subgroup : i in [1..#M1]];
  M2:=[M2[i]'subgroup : i in [1..#M2] | IsPrimitive(M2[i]'subgroup)];
  M1:=[H : H in M1 | #Core(G,H) eq 1 and IsPrimitive(H meet T)];
  M3:=[H : H in M2 | #Normaliser(G,H) eq #H];

```

```

M1:=M1 cat M3;

a1,a2,a3,a4:=RegTuplesPlus(G,M1,2,100);
b1,b2,b3,b4:=RegTuplesPlus(T,M2,2,100);
Append(~X,[a3,b3]);
[n,a1,b1];
end for;

```

The following code shows that $R_{\text{prim}}(G) = 4$ for every almost simple group $G \neq A_6, S_6$ with socle A_6 .

```

X:=[];
G:=AutomorphismGroupSimpleGroup("A6");
T:=Socle(G);
M2:=MaximalSubgroups(T);
M2:=[M2[i]'subgroup : i in [1..#M2] | IsPrimitive(M2[i]'subgroup)];
S:=LowIndexSubgroups(G,#G div #T);
for j in [2,3,5] do
  G:=S[j];
  M1:=MaximalSubgroups(G);
  M1:=[M1[i]'subgroup : i in [1..#M1]];
  M1:=[H : H in M1 | #Core(G,H) eq 1 and IsPrimitive(H meet T)];
  if j lt 5 then
    M3:=[H : H in M2 | #Normalizer(G,H) eq #H];
    M:=M1 cat M3;
  else
    M:=M1;
    for k in [2..4] do
      M3:=MaximalSubgroups(S[k]);
      M3:=[M3[i]'subgroup : i in [1..#M3]];
      M3:=[H : H in M3 | #Core(G,H) eq 1 and IsPrimitive(H meet T)];
      M:=M cat M3;
    end for;
  end if;

  a1,a2,a3,a4:=RegTuplesPlus(G,M,2,100);
  Append(~X,a3);
  [j,a1];
end for;

```

3.3. Imprimitve subgroups. Recall that $R_{\text{imprim}}(G)$ is the minimal integer k such that every k -tuple of imprimitive subgroups of G is regular.

Proposition 3.3. *Let $G = S_n$ or A_n with $6 \leq n \leq 30$ composite. Then $R_{\text{imprim}}(G) \leq 6$.*

Proof. We use the following MAGMA code to verify the result:

```

for n in [6..30] do
  if IsPrime(n) eq false then
    D:=Divisors(n);
    M:=[];
    for G in [Sym(n),Alt(n)] do
      M1:=[];
      for i in [2..#D-1] do
        a:=D[i];
        b:=n div a;
        Append(~M1,WreathProduct(Sym(a),Sym(b)) meet G);
      end for;
      Append(~M,M1);
    end for;

    a1,a2,a3,a4:=RegTuplesPlus(Sym(n),M[1],6,500);
    b1,b2,b3,b4:=RegTuplesPlus(Alt(n),M[2],6,500);
    [n,a1,b1];
  end if;
end for;

```

□

3.4. Soluble subgroups.

Proposition 3.4. *Suppose $G = S_n$ or A_n with $5 \leq n \leq 11$. Then $R_{\text{sol}}(G) \leq 5$, with equality if and only if $G = S_8$.*

Proof. We use the following code to compute $R_{\text{sol}}(G)$ in each case. This relies on the function `Sol`, which we defined in Section 1.3.

```
for n in [5..11] do
  G:=Sym(n);
  T:=Alt(n);
  A1:=[];
  A2:=[];
  a1,b1,c1:=Sol(G,100);
  a2,b2,c2:=Sol(T,100);
  Append(~A1,c1);
  Append(~A2,c2);
  n; [a1,b1]; [a2,b2]; " ";
end for;
```

For example, for $n = 5$ we get the output `[4,2], [3,6]`, which tells that $R_{\text{sol}}(S_5) = 4$ and $R_{\text{sol}}(A_5) = 3$, as well as the fact that S_5 has exactly 2 non-regular soluble triples, whereas A_5 has 6 non-regular soluble pairs (up to conjugacy and ordering). \square

Proposition 3.5. *Suppose $G = S_{16+m}$ and $H = (S_4 \wr S_4) \times S_m$, where $0 \leq m \leq 4$. Then $b(G, H) = 3$.*

Proof. For each m , the following code constructs G and H , and it uses random search to show that $b(G, H) \leq 3$. It then computes the number of (H, H) double cosets in G with size $|H|^2$, which is zero in every case, whence $b(G, H) = 3$ as claimed.

```
for m in [0..4] do
  G:=Sym(16+m);
  if m eq 0 then
    H:=WreathProduct(Sym(4),Sym(4));
  else
    H:=DirectProduct(WreathProduct(Sym(4),Sym(4)),Sym({17..16+m}));
  end if;
  repeat
    x1:=Random(G);
    x2:=Random(G);
  until
    #(H meet H^x1 meet H^x2) eq 1;
  R,S:=DoubleCosetRepresentatives(G,H,H);
  [m,#[s : s in S | s eq #H^2]];
end for;
```

\square

Proposition 3.6. *Suppose $G = S_n$ or A_n with $5 \leq n \leq 16$, and assume G has a soluble maximal subgroup. Then*

$$R_{\text{sol max}}(S_n) = \begin{cases} 5 & \text{if } n = 8 \\ 4 & \text{if } n = 5, 6 \\ 3 & \text{if } n = 7, 9, 12, 16 \\ 2 & \text{if } n = 11, 13 \end{cases}$$

and

$$R_{\text{sol max}}(A_n) = \begin{cases} 4 & \text{if } n = 8 \\ 3 & \text{if } n = 5, 6, 7, 9, 12, 16 \\ 2 & \text{if } n = 13 \end{cases}$$

Proof. Here we use the function `SolMax` (see Section 1.3). For n in the given range, note that S_n (respectively, A_n) has a soluble maximal subgroup unless $n \in \{10, 14, 15\}$ (respectively, $n \in \{10, 11, 14, 15\}$).

Note that “[0,0]” in the output signifies that G does not have a soluble maximal subgroup. In all other cases, “[k, l]” means that $R_{\text{sol max}}(G) = k$ and l is the number of non-regular soluble maximal $(k-1)$ -tuples, up to conjugacy and ordering (unless $k = 2$, in which case $l = 0$). For $R_{\text{sol max}}(G) = k \geq 3$, the procedure below also constructs all the non-regular soluble maximal $(k-1)$ -tuples, up to conjugacy and ordering.

```
for n in [5..16] do
  G:=Sym(n);
  T:=Alt(n);
  A1:=[];
  A2:=[];
  a1,b1,c1:=SolMax(G,100);
  a2,b2,c2:=SolMax(T,100);
  Append(~A1,c1);
  Append(~A2,c2);
  n; [a1,b1]; [a2,b2]; " ";
end for;
```

□

Proposition 3.7. *Let $G \neq S_6, A_6$ be an almost simple group with socle A_6 . Then*

$$R_{\text{sol max}}(G) = \begin{cases} 4 & \text{if } G = A_6.2^2 \\ 3 & \text{otherwise} \end{cases}$$

Proof. We use the following code to check this:

```
G:=AutomorphismGroupSimpleGroup("A6");
S:=LowIndexSubgroups(G,#G div #Socle(G));
for i in [2,3,5] do
  G:=S[i];
  A1:=[];
  a1,b1,c1:=SolMax(G,100);
  Append(~A1,c1);
  [i,a1,b1];
end for;
```

□

4. SPORADIC GROUPS

Let G be an almost simple sporadic group with socle T . The proof of [1, Theorem 2] relies entirely on computational methods. Here we detail all of the computations.

4.1. The regularity number. The regularity number of G is determined in [1, Proposition 4.3] and here we present the code we used to calculate this number. We also determine all the large $(k-1)$ -tuples (up to conjugacy and reordering), where $k = R(G)$; the latter information is recorded in the final column of [1, Table 3] (the term *large* tuple is defined in [1, Definition 4.1]). As in [1, Section 4.1], we proceed by dividing the possibilities for T into the following collections:

$$\begin{aligned} \mathcal{A}_1 &= \{M_{11}, M_{12}, M_{22}, M_{23}, M_{24}, J_1, J_2, J_3, \text{HS}, \text{McL}, \text{He}, \text{Suz}, \text{O}'\text{N}, \text{Ru}, \text{Co}_3, \text{Co}_2, \text{Fi}_{22}, \text{Fi}_{23}\} \\ \mathcal{A}_2 &= \{\text{Co}_1, \text{HN}, \text{Fi}'_{24}\} \\ \mathcal{A}_3 &= \{J_4, \text{Ly}, \text{Th}\} \\ \mathcal{A}_4 &= \{\mathbb{B}, \mathbb{M}\} \end{aligned}$$

Proposition 4.1. *The conclusion to [1, Proposition 4.3] holds if $T \in \mathcal{A}_1$.*

Proof. We use the following MAGMA code, which relies on the function `RegTuplesPlus` defined in Section 1.2. Note that the tuples `V1` and `V2` list the base numbers of the groups T and $T.2$, respectively, which can be read off from [5] (see [1, Proposition 4.2]).

```

U:=["M11","M12","M22","M23","M24","J1","J2","J3","HS","McL","He","Suz","ON","Ru","Co3","Co2","Fi22","Fi23"];
V1:=[4,5,5,6,7,3,4,3,4,5,4,4,3,4,6,6,5,5];
V2:=[0,5,5,0,0,0,4,3,5,5,4,4,3,0,0,0,6,0];

for u in [1..#U] do
  t:=U[u];
  G:=AutomorphismGroupSimpleGroup(t);
  T:=Socle(G);
  M2:=MaximalSubgroups(T);
  M2:=[M2[i]'subgroup : i in [1..#M2]];
  D:=[#M2[i] : i in [1..#M2]];
  ParallelSort(~D, ~M2);
  M2:=Reverse(M2);

  if #G ne #T then
    M1:=MaximalSubgroups(G);
    M1:=[M1[i]'subgroup : i in [1..#M1] | #Core(G,M1[i]'subgroup) eq 1];
    M3:=[H : H in M2 | #Normalizer(G,H) eq #H];
    for H in M3 do
      a:=exists(K){K : K in M1 | IsConjugate(G,K,H)};
      if a eq false then
        Append(~M1,H);
      end if;
    end for;
    D:=[#M1[i] : i in [1..#M1]];
    ParallelSort(~D, ~M1);
    M1:=Reverse(M1);
  end if;

  a1,a2,a3,a4:=RegTuplesPlus(T,M2,V1[u]-1,100);
  if #G ne #T then
    b1,b2,b3,b4:=RegTuplesPlus(G,M1,V2[u]-1,100);
  else
    b1:=0;
    b4:=[];
  end if;

  t;a1;a4; " "; b1;b4; " "; " ";
end for;

```

□

Proposition 4.2. *The conclusion to [1, Proposition 4.3] holds for $G = \text{Co}_1$.*

Proof. Here we need to show that $R(G) = 5$ and the only non-regular core-free 4-tuple is (H, H, H, H) , up to conjugacy, where $H = \text{Co}_2$. As explained in the proof of [1, Proposition 4.3], we begin by using the GAP function QGAP to determine all the maximal 4-tuples and 5-tuples τ with $\widehat{Q}(G, \tau) \geq 1$.

```

t:=CharacterTable("Co1");;
k:=3;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[2]; q[3];

```

This returns the following list of 4-tuples:

$$[1, 1, 1, 1], [1, 1, 1, 2], [1, 1, 1, 3], [1, 1, 1, 4], [1, 1, 1, 5], [1, 1, 1, 6] \quad (1)$$

and no 5-tuples are returned. As a consequence, we deduce that every core-free 5-tuple is regular and thus $R(G) \leq 5$. Since [5] gives $B(G) = 5$, we conclude that $R(G) = 5$.

The first tuple in (1) corresponds to (H, H, H, H) with $H = \text{Co}_2$, which we know is non-regular from [5]. Putting this tuple to one side, we switch to MAGMA and we use our function `RandomReg` to show that the remaining 4-tuples in (1) are all regular:

```
G:=AutomorphismGroupSimpleGroup("Co1");
M:=MaximalSubgroups(G);
M:=[M[i]'subgroup : i in [1..#M]];
A:=[#M[i] : i in [1..#M]];
ParallelSort(~A, ~M);
M:=Reverse(M);
for i in [1..#M] do [i,#M[i]]; end for;

A:=[[ 1, 1, 1, 2 ],[ 1, 1, 1, 3 ],[ 1, 1, 1, 4 ],[ 1, 1, 1, 5 ],[ 1, 1, 1, 6 ]];

R:=RandomReg(G,M,100,A);
R;
```

This returns an empty tuple and the proof is complete. □

Proposition 4.3. *The conclusion to [1, Proposition 4.3] holds for $T = \text{HN}$.*

Proof. First assume $G = T$. We use the function `QGAP` to determine all the core-free pairs and triples τ with $\widehat{Q}(G, \tau) \geq 1$.

```
t:=CharacterTable("HN");;
k:=2;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[1]; q[2];
```

This outputs

```
[ [ 1, 1 ], [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 1, 6 ], [ 1, 7 ],
  [ 2, 2 ], [ 2, 3 ], [ 2, 4 ], [ 2, 5 ], [ 2, 6 ], [ 2, 7 ], [ 3, 3 ],
  [ 1, 8 ], [ 1, 9 ], [ 1, 10 ], [ 1, 11 ], [ 1, 12 ], [ 1, 13 ], [ 1, 14 ],
  [ 2, 8 ], [ 2, 9 ], [ 2, 10 ], [ 2, 11 ], [ 2, 12 ], [ 2, 13 ], [ 3, 4 ],
  [ 3, 7 ], [ 3, 9 ], [ 4, 4 ], [ 4, 7 ], [ 4, 8 ], [ 7, 7 ], [ 7, 8 ],
  [ 8, 8 ] ]

[ ]
```

and thus every core-free triple for G is regular. In particular, $R(G) = 3$. We then switch to MAGMA to determine the regularity status of the pairs τ with $\widehat{Q}(G, \tau) \geq 1$:

```
G:=AutomorphismGroupSimpleGroup("HN");
G:=Socle(G);
M:=MaximalSubgroups(G);
M:=[M[i]'subgroup : i in [1..#M]];
A:=[#M[i] : i in [1..#M]];
ParallelSort(~A, ~M);
M:=Reverse(M);
for i in [1..#M] do [i,#M[i]]; end for;

A:=[[ 1, 1 ],[ 1, 2 ],[ 1, 3 ],[ 1, 4 ],[ 1, 5 ],[ 1, 6 ],[ 1, 7 ],[ 1, 8 ],[ 1, 9 ],[ 1, 10 ],
  [ 1, 11 ],[ 1, 12 ],[ 1, 13 ],[ 1, 14 ],[ 2, 2 ],[ 2, 3 ],[ 2, 4 ],[ 2, 5 ],[ 2, 6 ],[ 2, 7 ],
  [ 2, 8 ],[ 2, 9 ],[ 2, 10 ],[ 2, 11 ],[ 2, 12 ],[ 2, 13 ],[ 3, 3 ],[ 3, 4 ],[ 3, 7 ],[ 3, 9 ],
  [ 4, 4 ],[ 4, 7 ],[ 4, 8 ],[ 7, 7 ],[ 7, 8 ],[ 8, 8 ]];

R:=RegTuples(G,M,100,A);
Sort(R);
```

This outputs

```
[ [ 1, 1 ], [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 1, 6 ], [ 1, 7 ], [ 1, 8 ], [ 1, 9 ], [ 1, 10 ],
[ 1, 11 ], [ 1, 12 ], [ 2, 2 ], [ 2, 3 ], [ 2, 4 ], [ 2, 5 ], [ 2, 6 ], [ 2, 7 ], [ 2, 8 ], [ 2, 9 ],
[ 2, 10 ], [ 2, 11 ], [ 2, 12 ], [ 3, 3 ] ]
```

We now repeat the calculation for $G = \text{HN.2}$, working with tuples of subgroups in $\mathcal{M}'(G)$. First we use MAGMA to determine the maximal subgroups H of T with $N_G(H) = H$, up to conjugacy in G .

```
G:=AutomorphismGroupSimpleGroup("HN");
T:=Socle(G);
M:=MaximalSubgroups(T);
M:=[M[i]'subgroup : i in [1..#M]];
A:=[#M[i] : i in [1..#M]];
ParallelSort(~A, ~M);
M:=Reverse(M);
for i in [1..#M] do
  H:=M[i];
  if #Normaliser(G,H) eq #H then
    [i,#M[i]];
  end if;
end for;
```

This returns

```
[ 11, 190080 ], [ 12, 190080 ]
```

and we note that these two subgroups are conjugate in G (and isomorphic to $M_{12}.2$). We now carry out the GAP analysis on pairs and triples.

```
t:=CharacterTable("HN.2");;
t1:=CharacterTable("HN");;
k:=2;;
m:=[];;
a:=Maxes(t);;
a1:=Maxes(t1);;
for i in [2..Size(a)] do
  Add(m,a[i]);;
od;
Add(m,a1[11]);;

F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[1]; q[2];
```

This outputs

```
[ [ 1, 1 ], [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 1, 6 ], [ 1, 7 ], [ 1, 8 ], [ 1, 9 ], [ 1, 10 ],
[ 1, 11 ], [ 1, 12 ], [ 1, 13 ], [ 2, 2 ], [ 2, 3 ], [ 2, 4 ], [ 2, 5 ], [ 2, 6 ], [ 2, 7 ],
[ 2, 8 ], [ 2, 9 ], [ 2, 10 ], [ 2, 11 ], [ 2, 12 ], [ 2, 13 ], [ 3, 3 ], [ 3, 4 ], [ 3, 5 ],
[ 3, 7 ], [ 3, 8 ], [ 3, 9 ], [ 4, 4 ], [ 4, 5 ], [ 4, 6 ], [ 4, 7 ], [ 4, 8 ], [ 4, 9 ], [ 5, 7 ],
[ 7, 7 ], [ 7, 8 ], [ 7, 9 ], [ 8, 8 ] ]
```

```
[ ]
```

and thus $R(G) = 3$. As before, we now switch to MAGMA to determine the regularity status of the pairs τ with $\widehat{Q}(G, \tau) \geq 1$.


```

G:=AutomorphismGroupSimpleGroup("HN");
T:=Socle(G);
M:=MaximalSubgroups(G);
M:=[M[i]'subgroup : i in [1..#M] | #Core(G,M[i]'subgroup) eq 1];
A:=[#M[i] : i in [1..#M]];
ParallelSort(~A, ~M);
M:=Reverse(M);
M1:=MaximalSubgroups(T);
M1:=[M1[i]'subgroup : i in [1..#M1]];
A:=[#M1[i] : i in [1..#M1]];
ParallelSort(~A, ~M1);
M1:=Reverse(M1);
Append(~M,M1[11]);

for i in [1..#M] do
  [i,#M[i]];
end for;

A:=[[ 1, 1 ],[ 1, 2 ],[ 1, 3 ],[ 1, 4 ],[ 1, 5 ],[ 1, 6 ],[ 1, 7 ],[ 1, 8 ],[ 1, 9 ],[ 1, 10 ],
  [ 1, 11 ],[ 1, 12 ],[ 1, 13 ],[ 2, 2 ],[ 2, 3 ],[ 2, 4 ],[ 2, 5 ],[ 2, 6 ],[ 2, 7 ],
  [ 2, 8 ],[ 2, 9 ],[ 2, 10 ],[ 2, 11 ],[ 2, 12 ],[ 2, 13 ],[ 3, 3 ],[ 3, 4 ],[ 3, 5 ],
  [ 3, 7 ],[ 3, 8 ],[ 3, 9 ],[ 4, 4 ],[ 4, 5 ],[ 4, 6 ],[ 4, 7 ],[ 4, 8 ],[ 4, 9 ],[ 5, 7 ],
  [ 7, 7 ],[ 7, 8 ],[ 7, 9 ],[ 8, 8 ]];

R:=RegTuples(G,M,100,A);
Sort(R);

```

This returns

```

[[ 1, 1 ],[ 1, 2 ],[ 1, 3 ],[ 1, 4 ],[ 1, 5 ],[ 1, 6 ],[ 1, 7 ],[ 1, 8 ],[ 1, 9 ],[ 1, 10 ],
[ 1, 11 ],[ 1, 13 ],[ 2, 2 ],[ 2, 3 ],[ 2, 4 ],[ 2, 5 ],[ 2, 6 ],[ 2, 7 ],[ 2, 8 ],[ 2, 9 ],
[ 2, 10 ],[ 2, 13 ],[ 3, 3 ],[ 3, 4 ] ]

```

□

Proposition 4.4. *The conclusion to [1, Proposition 4.3] holds for $T = \text{Fi}'_{24}$.*

Proof. First assume $G = T$. We begin by applying QGAP to determine all the core-free 4-tuples and 5-tuples τ with $\widehat{Q}(G, \tau) \geq 1$:

```

t:=CharacterTable("Fi24'");;
k:=3;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[2]; q[3];

```

This outputs

```

[[ 1, 1, 1, 1 ],[ 1, 1, 1, 2 ],[ 1, 1, 1, 3 ],[ 1, 1, 1, 4 ],[ 1, 1, 1, 5 ],[ 1, 1, 1, 6 ] ]
[ 1, 1, 1, 1, 1 ] ]

```

It follows that $\tau = (H, H, H, H, H)$ is the only maximal 5-tuple with $\widehat{Q}(G, \tau) \geq 1$, where $H = \text{Fi}_{23}$, and we know that this is regular by [5]. Therefore, $R(G) = 5$ and it remains to determine all the non-regular maximal 4-tuples. To do this, we switch to MAGMA, noting that (H, H, H, H) is non-regular since $b(G, H) = 5$.

Given the above list of candidate non-regular maximal 4-tuples, we need to construct the following maximal subgroups of G :

$$1: \text{Fi}_{23}, \quad 2: 2.\text{Fi}_{22}.2, \quad 3: (3 \times \text{P}\Omega_8^+(3)).3.2, \quad 4: \Omega_{10}^-(2), \quad 5: 3^7.\Omega_7(3), \quad 6: 3^{1+10}:\text{U}_5(2).2$$

We do this using generators for the corresponding maximal subgroups of $T.2 = \text{Fi}_{24}$, which are available in the Web Atlas [10], noting that each subgroup H is of the form $M \cap T$, where M is a maximal subgroup of $T.2$.

```
G:=AutomorphismGroupSimpleGroup("Fi24");
T:=Socle(G);

[We construct the relevant maximal subgroups]

for i in [1..#M] do
  [i,#M[i]];
end for;
```

This returns

```
[ 1, 4089470473293004800 ]
[ 2, 258247006617600 ]
[ 3, 89139236659200 ]
[ 4, 25015379558400 ]
[ 5, 10028164124160 ]
[ 6, 4848782653440 ]
```

Next we observe that (H, H, H, K) is regular if and only if K has a trivial 3-point stabiliser in its action on G/H . The given permutation representation of G corresponds to the action of G on G/H , so this criterion is easy to check:

```
for i in [1..6] do
  H:=M[i];
  o:=Orbits(H);
  A:=[];
  for i1 in [1..#o] do
    H1:=Stabilizer(H,Representative(o[i1]));
    o1:=Orbits(H1);
    for i2 in [1..#o1] do
      H2:=Stabilizer(H1,Representative(o1[i2]));
      o2:=Orbits(H2);
      for i3 in [1..#o2] do
        H3:=Stabilizer(H2,Representative(o2[i3]));
        Append(~A,#H3);
      end for;
    end for;
  end for;
end for;
if 1 in A then
  w:=1;
else
  w:=0;
end if;
[i,w];
end for;
```

This outputs

```
[ 1, 0 ], [ 2, 0 ], [ 3, 1 ], [ 4, 1 ], [ 5, 1 ], [ 6, 1 ]
```

and we conclude that $(1, 1, 1, i)$ with $1 \leq i \leq 6$ is non-regular if and only if $i = 1$ or 2 .

Now assume $G = T.2$. First we need to determine the subgroups in $\mathcal{M}'(G)$. By inspecting [8, Table 1.1], we see that there are six classes of maximal subgroups H of T with $N_G(H) = H$, namely two classes for each of the following subgroups: $\text{He}.2$, $\text{U}_3(3).2$, $\text{L}_2(13).2$. In terms of [3] and the output of `Maxes` with respect to the character table of T , these subgroups are numbered 13,14,21,22,23,24. Note that the two T -classes of maximal subgroups isomorphic to $\text{He}.2$ are fused in G . And similarly for those of type $\text{U}_3(3).2$ and $\text{L}_2(13).2$.

```

t:=CharacterTable("Fi24");;
t0:=CharacterTable("Fi24'");;
k:=3;;
m:=[];;
a:=Maxes(t);;
a1:=Maxes(t0);;
for i in [2..Size(a)] do
  Add(m,a[i]);;
od;
for i in [13,21,23] do
  Add(m,a1[i]);;
od;

F:=[];;
for i in [1..Size(a)-1] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
f1:=FusionConjugacyClasses(t0,t);;
for i in [Size(a)..Size(a)+2] do
  t1:=CharacterTable(m[i]);;
  f:=FusionConjugacyClasses(t1,t0);;
  fus:=[];;
  for j in [1..Size(f)] do
    Add(fus,f1[f[j]]);;
  od;
  Add(F,fus);;
od;

q:=QGAP(t,m,F,k);;
q[2]; q[3];

```

This outputs

```

[ [ 1, 1, 1, 1 ], [ 1, 1, 1, 2 ], [ 1, 1, 1, 3 ], [ 1, 1, 1, 4 ],
  [ 1, 1, 1, 5 ], [ 1, 1, 1, 6 ], [ 1, 1, 1, 8 ] ]

[ [ 1, 1, 1, 1, 1 ] ]

```

Here the subgroup labelled by 1 is $H = \text{Fi}_{23} \times 2$ and by inspecting [5] we see that $(1, 1, 1, 1, 1)$ is regular, whereas $(1, 1, 1, 1)$ is non-regular. So it just remains to determine the regularity status of the tuples $(1, 1, 1, i)$ with $i \in \{2, 3, 4, 5, 6, 8\}$. To do this we switch to MAGMA and we construct the relevant maximal subgroups (using the generators in the Web Atlas).

```

G:=AutomorphismGroupSimpleGroup("Fi24");

[We construct the relevant maximal subgroups]

for i in [1..#M] do
  [i,#M[i]];
end for;

```

This returns (note that M[7] is the subgroup labelled 8 above, namely $(2 \times 2^2.U_6(2)):S_3$):

```

[ 1, 8178940946586009600 ]
[ 2, 516494013235200 ]
[ 3, 178278473318400 ]
[ 4, 50030759116800 ]
[ 5, 20056328248320 ]
[ 6, 9697565306880 ]
[ 7, 441447874560 ]

```

Note that the given permutation representation of G corresponds to the action of G on G/H . Moreover, a 4-tuple of the form (H, H, H, K) is regular if and only if K has a trivial 3-point stabiliser in its action on G/H .

```

for i in [1..7] do
  H:=M[i];
  o:=Orbits(H);
  A:=[];
  for i1 in [1..#o] do
    H1:=Stabilizer(H,Representative(o[i1]));
    o1:=Orbits(H1);
    for i2 in [1..#o1] do
      H2:=Stabilizer(H1,Representative(o1[i2]));
      o2:=Orbits(H2);
      for i3 in [1..#o2] do
        H3:=Stabilizer(H2,Representative(o2[i3]));
        Append(~A,#H3);
      end for;
    end for;
  end for;
  if 1 in A then
    w:=1;
  else
    w:=0;
  end if;
  [i,w];
end for;

```

This outputs

```
[ 1, 0 ], [ 2, 0 ], [ 3, 0 ], [ 4, 1 ], [ 5, 1 ], [ 6, 1 ], [ 7, 1 ]
```

and we conclude that $(1, 1, 1, i)$ with $i \in \{1, 2, 3, 4, 5, 6, 8\}$ is non-regular if and only if $i \in \{1, 2, 3\}$. \square

Proposition 4.5. *The conclusion to [1, Proposition 4.3] holds for $T = \text{Ly}$.*

Proof. As usual, we begin by applying QGAP to identify the maximal pairs and triples τ with $\widehat{Q}(G, \tau) \geq 1$:

```

t:=CharacterTable("Ly");;
k:=2;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[1]; q[2];

```

This outputs

```

[ [ 1, 1 ], [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 1, 6 ], [ 1, 7 ],
  [ 2, 2 ], [ 2, 3 ], [ 2, 4 ], [ 2, 5 ], [ 2, 6 ], [ 2, 7 ], [ 4, 4 ] ]

[ ]

```

and thus $R(G) = 3$. By inspecting [5] we deduce that the pairs $(1, 1)$ and $(2, 2)$ are non-regular, whereas $(4, 4)$ is regular. By comparing orders, we see that $(1, 2)$, $(1, 3)$, $(1, 4)$, $(1, 5)$, $(2, 3)$ and $(2, 4)$ are also non-regular. This leaves us to handle the following pairs:

$$(1, 6), (1, 7), (2, 5), (2, 6), (2, 7)$$

We now switch to MAGMA, constructing G and each of the relevant maximal subgroups in the matrix group $\mathrm{GL}_{111}(5)$. Random search then reveals that the pairs $(1, 7)$ and $(2, 7)$ are regular, but it is inconclusive in the three remaining cases.

```
G:=MatrixGroup("Ly",3);
a,M:=MaximalSubgroups(G,"Ly");
M:=[M[i]'group : i in [1..#M]];

repeat
  x:=Random(G);
until
  LMGOrder(M[1] meet M[7]^x) eq 1;

repeat
  x:=Random(G);
until
  LMGOrder(M[2] meet M[7]^x) eq 1;
```

Finally, we consider the remaining pairs (H, K) labelled $(1, 6)$, $(2, 5)$ and $(2, 6)$, with respect to the following maximal subgroups:

$$1: G_2(5), \quad 2: 3.\mathrm{McL}.2, \quad 5: 5^{1+4}.4S_6, \quad 6: 3^5:(2 \times M_{11})$$

We thank Derek Holt (personal communication) for resolving these three cases; using MAGMA, he was able to show that they are all non-regular. To do this, he considers the action of H on the module $V = \mathbb{F}_5^{111}$ for G , finding a d -dimensional submodule, where $d = 7$ for $H = G_2(5)$ and $d = 21$ for $H = 3.\mathrm{McL}.2$. The function `OrbitAction` is then used to construct the permutation representation of G on G/H and the conclusion follows by calculating all of the K -orbits on G/H . \square

Proposition 4.6. *The conclusion to [1, Proposition 4.3] holds for $T = \mathrm{Th}$.*

Proof. First we use QGAP to identify the maximal pairs and triples τ with $\widehat{Q}(G, \tau) \geq 1$:

```
t:=CharacterTable("Th");;
k:=2;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[1]; q[2];
```

This outputs

```
[ [ 1, 1 ], [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 2, 2 ], [ 2, 3 ], [ 3, 3 ] ]
[ ]
```

and so we conclude that $R(G) = 3$. By inspecting [5], we see that $(1, 1)$ and $(2, 2)$ are non-regular, whereas $(3, 3)$ is regular. By comparing orders, $(1, 2)$ is also non-regular and this leaves us to handle the following pairs:

$$(1, 3), (1, 4), (1, 5), (2, 3).$$

Switching to MAGMA, we can construct G and each relevant maximal subgroup inside $\mathrm{GL}_{248}(2)$ and then it is easy to check that the pairs $(1, 5)$ and $(2, 3)$ are regular:

```

G:=MatrixGroup("Th",1);
a,M:=MaximalSubgroups(G,"Th");
M:=[M[i]'group : i in [1..#M]];

repeat
  x:=Random(G);
until
  LMGOrder(M[1] meet M[5]^x) eq 1;

repeat
  x:=Random(G);
until
  LMGOrder(M[2] meet M[3]^x) eq 1;

```

We thank Derek Holt for showing that the remaining pairs (1, 3) and (1, 4) are both non-regular (personal communication). To do this, he proceeded as in the Lyons group case, working with MAGMA and the action of $H = {}^3D_4(2):3$ on the module \mathbb{F}_2^{248} to construct the permutation representation of G on G/H (which has degree 143127000). By calculating all of the orbits of K , he was able to conclude that K has no regular orbits, so the relevant pairs (H, K) are non-regular. \square

Proposition 4.7. *The conclusion to [1, Proposition 4.3] holds for $T = J_4$.*

Proof. As usual, we begin by applying the function QGAP to determine all the maximal pairs and triples τ with $\widehat{Q}(G, \tau) \geq 1$:

```

t:=CharacterTable("J4");;
k:=2;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[1]; q[2];

```

This outputs

```

[ [ 1, 1 ], [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 2, 2 ], [ 2, 3 ],
  [ 2, 4 ], [ 2, 5 ], [ 3, 3 ], [ 3, 4 ] ]

[ ]

```

and thus $R(G) = 3$. By inspecting [5], we see that (1, 1), (2, 2), and (3, 3) are non-regular. And by comparing orders, we deduce that (1, 2), (1, 3), (1, 4) and (2, 3) are also non-regular. This leaves the pairs

$$(1, 5), (2, 4), (2, 5), (3, 4)$$

We can use MAGMA to show that the latter two are regular, working with a matrix representation of G over \mathbb{F}_2 of dimension 112:

```

G:=MatrixGroup("J4",1);
a,M:=MaximalSubgroups(G,"J4");
M:=[M[i]'group : i in [1..#M]];

repeat
  x:=Random(G);
until
  LMGOrder(M[2] meet M[5]^x) eq 1;

repeat
  x:=Random(G);
until

```

```
LMGOrder(M[3] meet M[4]^x) eq 1;
```

This leaves the pairs $(1, 5)$ and $(2, 4)$, both of which we claim are non-regular. Here the first pair is (H, K) , where $H = 2^{11}.M_{24}$ and $K = U_3(11).2$. In this case, Derek Holt (personal communication) was able to construct G as a permutation group on G/H (as before, by considering an appropriate submodule with respect to the action of H on \mathbb{F}_2^{112}) and then calculate all of the K -orbits. However, this approach is not feasible for the pair $(2, 4)$, where we have

$$(H, K) = (2^{1+12}.3.M_{22}.2, 2^{3+12}.(S_5 \times L_3(2))),$$

due to the size of the degree $|G : H| = 3980549947$. We thank Jürgen Müller (personal communication) for resolving this case. To do this, he used the GAP package ORB [9] to establish the existence of sufficiently many distinct K -orbits on G/H to rule out the possibility of a regular orbit. Müller's computation was subsequently (and independently) verified by Derek Holt, using a different approach with MAGMA. \square

Proposition 4.8. *The conclusion to [1, Proposition 4.3] holds for $G = \mathbb{B}$.*

Proof. Here $B(G) = 4$ and so it suffices to show that every maximal 4-tuple τ is regular. To do this, we use our function QGAP to show that $\widehat{Q}(G, \tau) < 1$, noting that a small adjustment is required because the fusion map from H -classes to G -classes is not available in [3] when $H = (2^2 \times F_4(2)):2$. To get around this, we use the function PossibleClassFusions, which returns a list of 64 possible fusion maps. We find that the fixed point ratios $\text{fpr}(x, G/H)$ are independent of the choice of candidate fusion map, which then allows us to work with QGAP in the usual manner.

```
t:=CharacterTable("B");;
k:=2;;
m:=Maxes(t);;
F:=[];;
for i in [1..5] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
t1:=CharacterTable(m[6]);;
P:=PossibleClassFusions(t1,t);;
Add(F,P[1]);;
for i in [7..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[2]; q[3];
```

This outputs

```
[ [ 1, 1, 1 ], [ 1, 1, 2 ], [ 1, 1, 3 ], [ 1, 1, 4 ], [ 1, 1, 6 ], [ 1, 1, 7 ], [ 1, 2, 2 ] ]
[ ]
```

and we deduce that $R(G) = 4$. \square

Proposition 4.9. *The conclusion to [1, Proposition 4.3] holds for $G = \mathbb{M}$.*

Proof. First recall that G has 46 conjugacy classes of maximal subgroups. For 31 of these classes, the character table of a representative H and the fusion map from H -classes to G -classes is available in the GAP Character Table Library [3] via the function NamesOfFusionSources.

The relevant subgroups are as follows, and we use \mathcal{M}_1 to denote this collection of subgroups:

$2.\mathbb{B}$	$2^{1+24}.\text{Co}_1$	$3.\text{Fi}_{24}$	$2^2.{}^2E_6(2):S_3$
$3^{1+12}.2.\text{Suz}:2$	$S_3 \times \text{Th}$	$(D_{10} \times \text{HN}).2$	$5^{1+6}.2.\text{J}_2.4$
$(7:3 \times \text{He}):2$	$(A_5 \times A_{12}):2$	$5^{3+3}.(2 \times L_3(5))$	$(A_6)^3.(2 \times S_4)$
$(A_5 \times U_3(8):3):2$	$5^{2+2+4}:(S_3 \times \text{GL}_2(5))$	$(L_3(2) \times \text{Sp}_4(4):2).2$	$7^{1+4}:(3 \times 2S_7)$
$(5^2:[2^4] \times U_3(5)).S_3$	$(L_2(11) \times M_{12}):2$	$(A_7 \times (A_5 \times A_5):2^2):2$	$5^4:(3 \times 2.L_2(25)).2$
$7^{2+1+2}:\text{GL}_2(7)$	$M_{11} \times A_6.2^2$	$(S_5)^3:S_3$	$13^2.2.L_2(13).4$
$(7^2:(3 \times 2A_4) \times L_2(7)).2$	$(13:6 \times L_3(3)).2$	$13^{1+2}:(3 \times 4S_4)$	$L_2(71)$
$L_2(59)$	$L_2(41)$	$41:40$	

Let x_1, \dots, x_t be a complete set of representatives of the conjugacy classes in G of elements of prime order and set

$$a_i = |x_i^G|, \quad b_i = \max\{\text{fpr}(x_i, G/H) : H \in \mathcal{M}_1\} \quad (2)$$

for all $1 \leq i \leq t$. We can use GAP to calculate these numbers:

```
t:=CharacterTable("M");;
S:=SizesConjugacyClasses(t);;
O:=OrdersClassRepresentatives(t);;
s:=NamesOfFusionSources(t);;
B:=[36,29,6,39,10,31,34,11,12,13,14,15,16,17,18,35,19,20,21,22,23,24,25,26,27,28,38,9,8,7,33];;
for i in B do
  Print([i,Size(CharacterTable(s[i]))], " ", s[i], "\n");
od;

F:=[];
for i in B do
  t1:=CharacterTable(s[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;

a:=[];
b:=[];
P:=[];
for i in [1..Size(O)] do
  if IsPrime(O[i]) then
    Add(P,i);;
    Add(a,S[i]);;
  fi;
od;

S0:=[];
for i in B do
  t1:=CharacterTable(s[i]);;
  Add(S0,SizesConjugacyClasses(t1));;
od;

for i in P do
  G:=[];
  for j in [1..Size(B)] do
    fus:=F[j];;
    S1:=S0[j];;
    c:=0;
    for k in [1..Size(fus)] do
      if fus[k] = i then
        c:=c+S1[k];;
      fi;
    od;
    Add(G,c);;
  od;
  Add(b,Maximum(G)/S[i]);;
od;
```


Moreover, we can check that $\sum_i a_i b_i^3 < 1$, which implies that every triple of subgroups in \mathcal{M}_1 is regular.

```
z:=0;;
for i in [1..Size(a)] do
  z:=z+a[i]*b[i]^3;
od;
z < 1;
```

We now extend the analysis to maximal triples containing subgroups from the remaining classes of maximal subgroups. Let H be one of these subgroups and define

$$i_r(H) = |\{x \in H : |x| = r\}|, \quad c_r = \min\{a_i b_i : |x_i| = r\}$$

for each prime divisor r of $|G|$. Notice that if

$$i_r(H) \leq c_r \tag{3}$$

for all r , then $\text{fpr}(x_i, G/H) \leq b_i$ for all i . So by our previous calculation, if \mathcal{S} is a collection of subgroups of G such that (3) holds for all r and all $H \in \mathcal{S}$, then every triple of subgroups in \mathcal{S} is regular.

```
R:=PrimeDivisors(Size(t));;
A:=[];;
for r in R do
  B:=[];;
  for i in [1..Size(P)] do
    if 0[P[i]] = r then
      Add(B,a[i]*b[i]);;
    fi;
  od;
  Add(A,B);;
od;
C:=[];;
for i in [1..Size(A)] do
  Add(C,Minimum(A[i]));;
od;
R;
C;
```

This returns

```
[ 2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 41, 47, 59, 71 ]
```

```
[ 11734592583376, 500595349782528000, 9367743238695946498867200, 3954417208796381184000,
3147561728201838023619379200000, 447153330533129256960000, 61099727665094502811435008000000,
109336354769116478715199488000000, 90321336548400569373425664000000, 86565910978666325606400,
134025209071820199715405824000000, 1680, 88399605983540982791012352000000, 1740, 2520 ]
```

First consider the following 12 maximal subgroups of G :

$$\begin{aligned} & (\text{L}_2(11) \times \text{L}_2(11)).4, 11^2:(5 \times 2A_5), \text{U}_3(4):4, \text{L}_2(29):2, 7^2:\text{SL}_2(7), \text{L}_2(19):2, \text{L}_2(13):2 \\ & 3^{3+2+6+6}.\text{L}_3(3) \times \text{SD}_{16}, 3^{2+5+10}.\text{M}_{11} \times 2S_4, (3^2:2 \times \text{P}\Omega_8^+(3)).S_4 \\ & 2^{2+11+22}.\text{M}_{24} \times S_3, 2^{3+6+12+18}.\text{L}_3(2) \times 3S_6 \end{aligned}$$

In each of these cases, we can work with a permutation representation of H in the Web Atlas [10] in order to compute $i_r(H)$, which then allows us to verify the bound in (3).

For example, the following MAGMA code handles the case $H = (\text{L}_2(11) \times \text{L}_2(11)).4$:

```

R:=[ 2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 41, 47, 59, 71 ];

C:=[ 11734592583376, 500595349782528000, 9367743238695946498867200,
    3954417208796381184000, 3147561728201838023619379200000,
    447153330533129256960000, 61099727665094502811435008000000,
    109336354769116478715199488000000, 90321336548400569373425664000000,
    86565910978666325606400, 134025209071820199715405824000000, 1680,
    88399605983540982791012352000000, 1740, 2520 ];

H<x,y>:=PermutationGroup<24|\[13,14,15,16,17,18,19,20,21,22,23,24,10,9,8,7,6,5,4,3,2,1,11,12],
\[10,5,7,8,2,9,3,4,6,1,12,11,14,15,16,17,18,19,20,21,22,23,13,24]>;

cl:=Classes(H);
A:=[];
for j in [1..#R] do
  r:=R[j];
  z:=0;
  for i in [2..#cl] do
    if cl[i][1] eq r then
      z:=z+cl[i][2];
    end if;
  end for;
  if z gt C[j] then
    Append(~A,[r,z,C[j]]);
  end if;
end for;
A;

```

This returns an empty tuple, which means that (3) holds for every prime divisor r of $|G|$.

Next assume H is one of the following maximal subgroups:

$$2^{5+10+20} \cdot (S_3 \times L_5(2)), \quad 2^{10+16} \cdot \Omega_{10}^+(2)$$

Here an upper bound on $i_r(H)$ is given in [5, Proposition 3.8] and one checks that the inequality in (3) holds for all r .

Finally, it just remains to consider the maximal subgroup $H = 3^8 \cdot \text{P}\Omega_8^-(3) \cdot 2$. Here the Web Atlas provides a permutation representation of H on 805896 points and as before we can use this to compute $i_r(H)$ for every prime divisor r of $|G|$. In this way, we find that the inequality in (3) holds unless $r = 41$. More precisely, G has a unique class of elements of order 41 and we get

$$i_{41}(H) = 16245625881139200 > c_{41} = 1680.$$

However, if we now redefine b_i to be $i_{41}(H)/a_i$ (for the unique i such that $|x_i| = 41$) then it is easy to check that the inequality $\sum_i a_i b_i^3 < 1$ is still satisfied.

We conclude that every maximal triple is regular, which completes the proof for $G = \mathbb{M}$. \square

4.2. Soluble triples. As before, let G be an almost simple sporadic group with socle T . In this section, we present full details of the proof of [1, Proposition 4.12], which establishes the bound $R_{\text{sol}}(G) \leq 3$. As in [1, Section 4.2], we divide the possibilities for T into the following collections:

$$\begin{aligned} \mathcal{B}_1 &= \{M_{11}, M_{12}, M_{22}, M_{23}, M_{24}, J_1, J_2, J_3, \text{HS}, \text{McL}, \text{He}, \text{Suz}, \text{Ru}, \text{Co}_3, \text{O}'\text{N}\} \\ \mathcal{B}_2 &= \{\text{Co}_2, \text{Fi}_{22}, \text{Fi}_{23}, \text{HN}, J_4, \text{Ly}, \text{Th}, \mathbb{M}\} \\ \mathcal{B}_3 &= \{\text{Co}_1, \text{Fi}'_{24}\} \\ \mathcal{B}_4 &= \{\mathbb{B}\} \end{aligned}$$

Proposition 4.10. *The conclusion to [1, Proposition 4.12] holds if $T \in \mathcal{B}_1$.*

Proof. We apply the MAGMA function `Sol` from Section 1.3:

```

U:=["M11","M12","M22","M23","M24","J1","J2","J3","HS","McL","He","Suz","Ru","Co3","ON"];

for t in U do
  G:=AutomorphismGroupSimpleGroup(t);
  T:=Socle(G);
  a1,b1,c1:=Sol(G,100);
  if #G eq #T then
    t:[a1,b1]; " ";
  else
    a2,b2,c2:=Sol(T,100);
    t:[a2,b2];[a1,b1]; " ";
  end if;
end for;

```

□

Proposition 4.11. *The conclusion to [1, Proposition 4.12] holds if $T \in \mathcal{B}_2$.*

Proof. As explained in the proof of [1, Lemma 4.16], we may assume $T \in \{\text{Co}_2, \text{Fi}_{22}, \text{Fi}_{23}\}$ and our goal is to show that every soluble triple is regular. First we handle the groups with $G = T$. The following code uses the function `MaxSolubleSubgps` from Section 1.3 to construct the subgroup collections $\mathcal{S}_3(H)$ as H runs through a set of representatives of the conjugacy classes of maximal subgroups of G (see [1, Definition 4.14]).

```

U:=["Co2","Fi22","Fi23"];

for t in U do
  G:=AutomorphismGroupSimpleGroup(t);
  G:=Socle(G);
  M:=MaximalSubgroups(G);
  M:=[M[i]'subgroup : i in [1..#M]];
  D:=[#M[i] : i in [1..#M]];
  ParallelSort(~D, ~M);
  M:=Reverse(M);

  S:=[];
  SO:=[];
  for i in [1..#M] do
    S1:=[];
    A:=MaxSolubleSubgps(G,M[i],3);
    for H in A do
      Append(~S1,H);
      a:=exists(K){K : K in S | IsConjugate(G,K,H)};
      if a eq false then
        Append(~S,H);
      end if;
    end for;
    Append(~SO,S1);
  end for;
  D:=[#S[i] : i in [1..#S]];
  ParallelSort(~D, ~S);
  S:=Reverse(S);

  A1:=CartesianPower([1..#M],2);
  A1:=[[x[i] : i in [1..2]] : x in A1];
  A1:=[x : x in A1 | x eq Sort(x)];

  A:=[];
  g:=#G;
  for a in A1 do
    for k in [1..#S] do
      b:=[M[a[1]],M[a[2]],S[k]];
      if &*[#b[i] : i in [1..3]] gt g^2 then
        Append(~A,a);
        break k;
      else
        w:=0;
      end if;
    end for;
  end for;

```

```

for i in [1..50] do
  x1:=Random(G);
  x2:=Random(G);
  if #(b[1] meet b[2]^x1 meet b[3]^x2) eq 1 then
    w:=1;
    break i;
  end if;
end for;
if w eq 0 then
  Append(~A,a);
  break k;
end if;
end if;
end for;
end for;

if #A eq 0 then
  B:=[];
else
  B:=[];
  for k in [1..#A] do
    a:=A[k];
    b:=a;
    A1:=S0[a[1]];
    A2:=S0[a[2]];
    [k,#A,#B]; a; [#A1,#A2]; " ";
    for j in [a[2]..#M] do
      H1:=M[j];
      for H2 in A1 do
        for H3 in A2 do
          w:=0;
          for i in [1..100] do
            x1:=Random(G);
            x2:=Random(G);
            if #(H1 meet H2^x1 meet H3^x2) eq 1 then
              w:=1;
              break i;
            end if;
          end for;
          if w eq 0 then
            Append(~b,j);
            Append(~B,b);
            break j;
          end if;
        end for;
      end for;
    end for;
  end for;
end if;
t; B;
end for;

```

For $G = \text{Fi}_{23}$, this returned $[[2,2,2]]$. We use the following procedure to settle this case:

```

G:=AutomorphismGroupSimpleGroup("Fi23");
M:=MaximalSubgroups(G);
M:=[M[i]'subgroup : i in [1..#M]];
D:=[#M[i] : i in [1..#M]];
ParallelSort(~D, ~M);
M:=Reverse(M);
S:=MaxSolubleSubgps(G,M[2],4);
for i in [1..#S] do
  for j in [i..#S] do
    repeat
      x1:=Random(G);
      x2:=Random(G);

```

```

    until #(M[2] meet S[i]^x1 meet S[j]^x2) eq 1;
  end for;
end for;

```

It remains to handle the case $G = \text{Fi}_{22}.2$. Recall that our goal is to show that $R_{\text{sol}}(G) = 3$ and by the above procedure we know that $R_{\text{sol}}(T) = 3$. Therefore, [1, Lemma 2.11] implies that any soluble triple (H, K, L) with at least one component in T is regular. We now proceed as above, excluding any triples containing a subgroup in T .

```

G:=AutomorphismGroupSimpleGroup("Fi22");
T:=Socle(G);
M:=MaximalSubgroups(G);
M:=[M[i]'subgroup : i in [1..#M] | #Core(G,M[i]'subgroup) eq 1];
D:=[#M[i] : i in [1..#M]];
ParallelSort(~D, ~M);
M:=Reverse(M);

```

```

S:=[];
S0:=[];
for i in [1..#M] do
  S1:=[];
  A:=MaxSolubleSubgps(G,M[i],4);
  for H in A do
    if #(H meet T) ne #H then
      Append(~S1,H);
      a:=exists(K){K : K in S | IsConjugate(G,K,H)};
      if a eq false then
        Append(~S,H);
      end if;
    end if;
  end for;
  Append(~S0,S1);
end for;
D:=[#S[i] : i in [1..#S]];
ParallelSort(~D, ~S);
S:=Reverse(S);

```

```

A1:=CartesianPower([1..#M],2);
A1:=[x[i] : i in [1..2]] : x in A1];
A1:=x : x in A1 | x eq Sort(x)];

```

```

A:=[];
g:=#G;
for a in A1 do
  for k in [1..#S] do
    b:=[M[a[1]],M[a[2]],S[k]];
    if &*[#b[i] : i in [1..3]] gt g^2 then
      Append(~A,a);
      break k;
    else
      w:=0;
      for i in [1..50] do
        x1:=Random(G);
        x2:=Random(G);
        if #(b[1] meet b[2]^x1 meet b[3]^x2) eq 1 then
          w:=1;
          break i;
        end if;
      end for;
      if w eq 0 then
        Append(~A,a);
        break k;
      end if;
    end if;
  end for;
end for;
end for;

```

```

if #A eq 0 then
  B:=[];
else
  B:=[];
  for k in [1..#A] do
    a:=A[k];
    b:=a;
    A1:=SO[a[1]];
    A2:=SO[a[2]];
    [k,#A,#B]; a; [#A1,#A2]; " ";
    for j in [a[2]..#M] do
      H1:=M[j];
      for H2 in A1 do
        for H3 in A2 do
          w:=0;
          for i in [1..100] do
            x1:=Random(G);
            x2:=Random(G);
            if #(H1 meet H2^x1 meet H3^x2) eq 1 then
              w:=1;
              break i;
            end if;
          end for;
          if w eq 0 then
            Append(~b,j);
            Append(~B,b);
            break j;
          end if;
        end for;
      end for;
    end for;
  end for;
end if;
B;

```

This returns $[[1, 1, 1]]$. So to complete the argument for $G = \text{Fi}_{22}.2$ it suffices to show that every triple of the form (H_1, H_2, H_3) with each H_i contained in $\mathcal{S}_3(H)$ is regular, where $H = 2.U_6(2).2$ is the largest core-free maximal subgroup of G . The following procedure carries out this computation.

```

G:=AutomorphismGroupSimpleGroup("Fi22");
T:=Socle(G);
M:=MaximalSubgroups(G);
M:=[M[i]'subgroup : i in [1..#M] | #Core(G,M[i]'subgroup) eq 1];
D:=[#M[i] : i in [1..#M]];
ParallelSort(~D, ~M);
M:=Reverse(M);

S:=[];
H:=M[1];
A:=MaxSolubleSubgps(G,H,3);
for H in A do
  if #(H meet T) ne #H then
    Append(~S,H);
  end if;
end for;

B:=[];
for i in [1..#S] do
  for j in [i..#S] do
    for k in [j..#S] do
      repeat
        x1:=Random(G);
        x2:=Random(G);
      until
        #(S[i] meet S[j]^x1 meet S[k]^x2) eq 1;
    end for;
  end for;
end for;

```

```

end for;
end for;

```

□

Proposition 4.12. *The conclusion to [1, Proposition 4.12] holds if $T \in \mathcal{B}_3$.*

Proof. First assume $G = \text{Fi}'_{24}$. Here the following GAP code shows that every non-regular maximal triple for G has an Fi_{23} component, whence $R_{\text{sol}}(G) \leq 3$ by [1, Lemma 2.11] since we have already shown that $R_{\text{sol}}(\text{Fi}_{23}) = 3$.

```

t:=CharacterTable("Fi24'");;
k:=2;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[2];

```

This returns

```

[ [ 1, 1, 1 ], [ 1, 1, 2 ], [ 1, 1, 3 ], [ 1, 1, 4 ], [ 1, 1, 5 ], [ 1, 1, 6 ], [ 1, 1, 7 ],
  [ 1, 1, 8 ], [ 1, 1, 9 ], [ 1, 1, 10 ], [ 1, 1, 11 ], [ 1, 1, 12 ], [ 1, 1, 13 ],
  [ 1, 1, 14 ], [ 1, 1, 15 ], [ 1, 1, 16 ], [ 1, 2, 2 ], [ 1, 2, 3 ], [ 1, 2, 4 ],
  [ 1, 2, 5 ], [ 1, 2, 6 ], [ 1, 3, 3 ], [ 1, 3, 4 ], [ 1, 3, 5 ], [ 1, 3, 6 ], [ 1, 4, 4 ] ]

```

and the result follows since “1” is the label for Fi_{23} .

Next assume $G = \text{Co}_1$. The smaller Conway groups $H = \text{Co}_2$ and Co_3 arise as maximal subgroups of G and we have already shown that $R_{\text{sol}}(H) \leq 3$. Therefore, every soluble triple with a component subgroup in Co_2 or Co_3 is regular. We begin by using GAP to identify all the maximal triples σ with $\widehat{Q}(G, \sigma) \geq 1$, excluding any triples with a Co_2 or Co_3 component:

```

t:=CharacterTable("Co1");;
k:=2;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
for a in q[2] do
  if (1 in a) = false and (4 in a) = false then
    Add(A,a);;
  fi;
od;
A;

```

This returns

```

[ 2, 2, 2 ], [ 2, 2, 3 ], [ 2, 2, 5 ], [ 2, 2, 6 ], [ 2, 2, 7 ], [ 2, 3, 3 ], [ 2, 3, 5 ],
  [ 2, 3, 6 ], [ 2, 5, 5 ], [ 3, 3, 3 ], [ 3, 3, 5 ], [ 3, 3, 6 ]

```

Next we switch to MAGMA and we construct G and each of the 6 relevant maximal subgroups inside S_n , where $n = 98280$. We then apply random search to show that 10 of the 12 maximal triples above are regular.

```

G:=AutomorphismGroupSimpleGroup("Co1");
M:=MaximalSubgroups(G);
M:=[M[i]'subgroup : i in [1..#M]];
A:=[#M[i] : i in [1..#M]];
ParallelSort(~A, ~M);
M:=Reverse(M);
for i in [1..#M] do [i,#M[i]]; end for;

A:=[ [ 2, 2, 2 ], [ 2, 2, 3 ], [ 2, 2, 5 ], [ 2, 2, 6 ], [ 2, 2, 7 ], [ 2, 3, 3 ], [ 2, 3, 5 ],
      [ 2, 3, 6 ], [ 2, 5, 5 ], [ 3, 3, 3 ], [ 3, 3, 5 ], [ 3, 3, 6 ] ];

R:=RandomReg(G,M,100,A);
R;

```

This returns:

```
[ [ 2, 2, 2 ], [ 2, 2, 3 ] ]
```

Finally, we proceed as follows to show that $R_{\text{sol}}(G) \leq 3$:

```

S:=MaxSolubleSubgps(G,M[2],3);
for j in [2,3] do
  H1:=M[j];
  H2:=M[2];
  for i in [1..#S] do
    H3:=S[i];
    repeat
      x1:=Random(G);
      x2:=Random(G);
    until
      #(H1 meet H2^x1 meet H3^x2) eq 1;
  end for;
end for;

```

To complete the proof, let us assume $G = \text{Fi}_{24} = T.2$. We know that $R_{\text{sol}}(T) \leq 3$, so we can ignore all the soluble triples with at least one component contained in T . We begin by determining all of the core-free maximal triples σ with $\widehat{Q}(G, \sigma) \geq 1$:

```

t:=CharacterTable("Fi24");;
k:=2;;
m:=Maxes(t);;
F:=[];;
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
A:=[];;
for a in q[2] do
  if (1 in a) = false then
    Add(A,a);;
  fi;
od;
A;

```

This outputs

```

[ [ 2, 2, 2 ], [ 2, 2, 3 ], [ 2, 2, 4 ], [ 2, 2, 5 ], [ 2, 2, 6 ],
  [ 2, 2, 7 ], [ 2, 2, 8 ], [ 2, 2, 9 ], [ 2, 2, 10 ], [ 2, 2, 11 ],
  [ 2, 2, 12 ], [ 2, 2, 13 ], [ 2, 2, 14 ], [ 2, 2, 15 ], [ 2, 2, 16 ],
  [ 2, 3, 3 ], [ 2, 3, 4 ], [ 2, 3, 5 ], [ 2, 3, 6 ], [ 2, 3, 7 ],
  [ 2, 3, 9 ], [ 2, 3, 10 ], [ 2, 4, 4 ], [ 2, 4, 5 ], [ 2, 4, 6 ],
  [ 2, 4, 7 ], [ 2, 4, 9 ], [ 2, 5, 5 ], [ 2, 5, 6 ], [ 2, 5, 7 ],
  [ 2, 6, 6 ], [ 2, 6, 7 ], [ 2, 7, 7 ] ]

```


We switch to MAGMA to handle these cases. To do this, we use the generators in the Web Atlas [10] to construct a representative of each conjugacy class of maximal subgroups of G .

```
G:=AutomorphismGroupSimpleGroup("Fi24");
M:=[Socle(G)];

[We construct the relevant maximal subgroups]

for i in [1..#M] do
  [i,#M[i]];
end for;
```

This returns

```
[ 1, 1255205709190661721292800 ]
[ 2, 8178940946586009600 ]
[ 3, 516494013235200 ]
[ 4, 178278473318400 ]
[ 5, 50030759116800 ]
[ 6, 20056328248320 ]
[ 7, 9697565306880 ]
[ 8, 1002795171840 ]
[ 9, 441447874560 ]
[ 10, 321052999680 ]
[ 11, 35814871872 ]
[ 12, 27549901440 ]
[ 13, 25082265600 ]
[ 14, 3963617280 ]
[ 15, 3963617280 ]
[ 16, 305690112 ]
```

Next, we use random search to further reduce the list of maximal triples we need to consider.

```
A:=[ [ 2, 2, 2 ], [ 2, 2, 3 ], [ 2, 2, 4 ], [ 2, 2, 5 ], [ 2, 2, 6 ],
      [ 2, 2, 7 ], [ 2, 2, 8 ], [ 2, 2, 9 ], [ 2, 2, 10 ], [ 2, 2, 11 ],
      [ 2, 2, 12 ], [ 2, 2, 13 ], [ 2, 2, 14 ], [ 2, 2, 15 ], [ 2, 2, 16 ],
      [ 2, 3, 3 ], [ 2, 3, 4 ], [ 2, 3, 5 ], [ 2, 3, 6 ], [ 2, 3, 7 ],
      [ 2, 3, 9 ], [ 2, 3, 10 ], [ 2, 4, 4 ], [ 2, 4, 5 ], [ 2, 4, 6 ],
      [ 2, 4, 7 ], [ 2, 4, 9 ], [ 2, 5, 5 ], [ 2, 5, 6 ], [ 2, 5, 7 ],
      [ 2, 6, 6 ], [ 2, 6, 7 ], [ 2, 7, 7 ] ];

g:=#G;
B1:=[];
B2:=[];
for i in [1..#A] do
  [i,#B1,#B2];
  a:=A[i];
  if &*[#M[a[i]] : i in [1..3]] ge g^2 then
    Append(~B1,a);
  else
    w:=0;
    for j in [1..10] do
      x1:=Random(G);
      x2:=Random(G);
      if #(M[a[1]] meet M[a[2]]^x1 meet M[a[3]]^x2) eq 1 then
        w:=1;
        break j;
      end if;
    end for;
    if w eq 0 then
      Append(~B2,a);
    end if;
  end if;
end for;
B:=B1 cat B2;
B;
```

This returns

```
[
  [ 2, 2, 2 ], [ 2, 2, 3 ], [ 2, 2, 4 ], [ 2, 2, 5 ], [ 2, 2, 6 ], [ 2, 2, 7 ], [ 2, 2, 8 ],
  [ 2, 2, 9 ], [ 2, 2, 10 ], [ 2, 2, 11 ], [ 2, 2, 12 ], [ 2, 2, 13 ], [ 2, 3, 3 ],
  [ 2, 3, 4 ], [ 2, 3, 5 ], [ 2, 3, 6 ], [ 2, 3, 7 ], [ 2, 4, 4 ]
]
```

and we further reduce this list as follows:

```
S:=MaxSolubleSubgps(G,M[2],3);
S:=[H : H in S | #(H meet M[1]) ne #H];

A:=[ [ 2, 2, 2 ], [ 2, 2, 3 ], [ 2, 2, 4 ], [ 2, 2, 5 ], [ 2, 2, 6 ], [ 2, 2, 7 ], [ 2, 2, 8 ],
      [ 2, 2, 9 ], [ 2, 2, 10 ], [ 2, 2, 11 ], [ 2, 2, 12 ], [ 2, 2, 13 ], [ 2, 3, 3 ],
      [ 2, 3, 4 ], [ 2, 3, 5 ], [ 2, 3, 6 ], [ 2, 3, 7 ], [ 2, 4, 4 ] ];

B:=[];
for i in [1..#A] do
  a:=A[i];
  H2:=M[a[2]];
  H3:=M[a[3]];
  for j in [1..#S] do
    H1:=S[j];
    w:=0;
    for k in [1..10] do
      x1:=Random(G);
      x2:=Random(G);
      if #(H1 meet H2^x1 meet H3^x2) eq 1 then
        w:=1;
        break k;
      end if;
    end for;
    if w eq 0 then
      Append(~B,a);
      break j;
    end if;
  end for;
end for;
B;
```

This returns

```
[ [ 2, 2, 2 ], [ 2, 2, 3 ], [ 2, 2, 4 ], [ 2, 2, 5 ] ]
```

and we complete the proof as follows:

```
A:= [ [ 2, 2, 2 ], [ 2, 2, 3 ], [ 2, 2, 4 ], [ 2, 2, 5 ] ];
for i in [1..#A] do
  a:=A[i];
  H3:=M[a[3]];
  for j in [1..#S] do
    for k in [j..#S] do
      H1:=S[j];
      H2:=S[k];
      w:=0;
      repeat
        x1:=Random(G);
        x2:=Random(G);
      until #(H1 meet H2^x1 meet H3^x2) eq 1;
    end for;
  end for;
end for;
```

□

Proposition 4.13. *The conclusion to [1, Proposition 4.12] holds for all groups G with socle $T \in \mathcal{B}_4$.*

Proof. Here $G = \mathbb{B}$ is the Baby Monster. Let x_1, \dots, x_t be a complete set of representatives of the conjugacy classes of elements of prime order in G . Using the GAP Character Table Library [3], we start by computing

$$a_i = |x_i^G|, \quad b_i = \max\{\text{fpr}(x_i, G/H) : H \in \mathcal{M}_0\}$$

for each i , where \mathcal{M}_0 is a set of representatives of the conjugacy classes of maximal subgroups of G , excluding $2.^2E_6(2):2$. In addition, we check that

$$\sum_{i=1}^t a_i b_i^3 < 1,$$

which implies that every maximal triple for G is regular, so long as it does not contain a component conjugate to $2.^2E_6(2):2$.

```
t:=CharacterTable("B");;
S:=SizesConjugacyClasses(t);;
O:=OrdersClassRepresentatives(t);;
m:=Maxes(t);;
F:=[];
for i in [1..5] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;
t1:=CharacterTable(m[6]);;
P:=PossibleClassFusions(t1,t);;
Add(F,P[1]);;
for i in [7..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(F,FusionConjugacyClasses(t1,t));;
od;

a:=[];
b:=[];
P:=[];
Q:=[];
for i in [1..Size(O)] do
  if IsPrime(O[i]) then
    Add(P,i);;
    Add(a,S[i]);;
    Add(Q,O[i]);;
  fi;
od;

S0:=[];
for i in [1..Size(m)] do
  t1:=CharacterTable(m[i]);;
  Add(S0,SizesConjugacyClasses(t1));;
od;

for i in P do
  G:=[];
  for j in [2..Size(m)] do
    fus:=F[j];;
    S1:=S0[j];;
    c:=0;
    for k in [1..Size(fus)] do
      if fus[k] = i then
        c:=c+S1[k];;
      fi;
    od;
    Add(G,c);;
  od;
od;
```

```

Add(b,Maximum(G)/S[i]);
od;

z:=0;
for i in [1..Size(a)] do
  z:=z+a[i]*b[i]^3;
od;
z < 1;

```

Set $L = 2.^2E_6(2):2$ and write $\mathcal{M} = \mathcal{M}_1 \cup \mathcal{M}_2$, where

$$\mathcal{M}_1 = \{\text{maximal subgroups of } L, \text{ other than } 2.^2E_6(2)\}$$

$$\mathcal{M}_2 = \{\text{maximal subgroups of } 2.^2E_6(2) < L\}$$

It suffices to show that

$$\text{fpr}(x_i, G/H) \leq b_i$$

for all $1 \leq i \leq t$ and all $H \in \mathcal{M}$. We use MAGMA to do this.

```
P:=[ 2, 2, 2, 2, 3, 3, 5, 5, 7, 11, 13, 17, 19, 23, 23, 31, 31, 47, 47 ];
```

```
a:=[ 13571955000, 11707448673375, 156849238149120000, 355438141723665000,
  10725600877608960000, 63622675428068556800000, 9367743238695946498867200,
  692463580204404365196263424, 7360369687546815106252800000,
  3147561728201838023619379200000, 13316607311623160869158912000000,
  61099727665094502811435008000000, 109336354769116478715199488000000,
  90321336548400569373425664000000, 90321336548400569373425664000000,
  134025209071820199715405824000000, 134025209071820199715405824000000,
  88399605983540982791012352000000, 88399605983540982791012352000000 ];
```

```
b:=[ 793/2950425, 269689951/11707448673375, 13/33634845, 303281/780496578225,
  11/105887475, 2/5353200125, 11/22299902235, 2/93659589387, 1/37166503725,
  1/780496578225, 1/126996316160000, 1/253992632320000, 1/22892381208576000,
  1/11707448673375, 1/11707448673375, 1/45784762417152000,
  1/45784762417152000, 1/3843461129719173164826624000000,
  1/3843461129719173164826624000000 ];
```

As explained in the proof of [1, Lemma 4.18], we only need to consider the subgroups $H \in \mathcal{M}_1$. Let Z be the centre of L and write $\bar{H} = H/Z$, which is a maximal subgroup of the almost simple group $\bar{L} = L/Z = 2.^2E_6(2):2$.

We begin with $\bar{H} = (L_3(2) \times L_3(4):2):2$:

```
G:=DirectProduct(PSL(3,2),PSL(3,4));
A:=AutomorphismGroup(G);
A:=PermutationGroup(A);
h:=#A div 3;
Q:=PrimeFactors(h);
B:=[];
for i in [1..#a] do
  if (P[i] in Q) and (h/a[i] gt b[i]) then
    Append(~B,i);
  end if;
end for;
B;
cl:=Classes(A);
z:=&+[cl[i][2] : i in [1..#cl] | cl[i][1] eq 2];
(2*z+1)/a[1] le b[1];
```

Next we handle $\bar{H} = 3^{1+6}:2^{3+6}:3^2:2^2$, followed by $\bar{H} = S_3 \times U_6(2):2$, $S_3 \times \Omega_8^+(2):S_3$, $U_3(8):6$ and $U_3(2):2 \times G_2(2)$:

```
h:=3^7*2^9*3^2*2^2*2;
Q:=[2,3];
B:=[];
```

```

for i in [1..#a] do
  if (P[i] in Q) and (h/a[i] gt b[i]) then
    Append(~B,i);
  end if;
end for;
B;

A:=[];
G:=AutomorphismGroupSimpleGroup("U",6,2);
S:=LowIndexSubgroups(G,#G div #Socle(G));
Append(~A,DirectProduct(Sym(3),S[2]));

G:=AutomorphismGroupSimpleGroup("O+",8,2);
Append(~A,DirectProduct(Sym(3),G));

G:=AutomorphismGroupSimpleGroup("U",3,8);
S:=LowIndexSubgroups(G,#G div #Socle(G));
Append(~A,S[7]);

G:=PSU(3,2);
L:=AutomorphismGroup(G);
L:=PermutationGroup(L);
S:=LowIndexSubgroups(L,#L div #G);
G1:=S[5];
G2:=AutomorphismGroupSimpleGroup("U",3,3);
Append(~A,DirectProduct(G1,G2));

for j in [1..#A] do
  H:=A[j];
  h:=2*#H;
  Q:=PrimeFactors(h);
  cl:=Classes(H);
  C:=[];
  for q in Q do
    if q eq 2 then
      Append(~C,2*#cl : i in [1..#cl] | cl[i][1] eq q)+1);
    else
      Append(~C,#+cl[i][2] : i in [1..#cl] | cl[i][1] eq q);
    end if;
  end for;
  B:=[];
  for i in [1..#a] do
    c:=exists(k){k : k in [1..#Q] | Q[k] eq P[i]};
    if c eq true then
      if (C[k]/a[i] gt b[i]) then
        Append(~B,i);
      end if;
    end if;
  end for;
  j; B; " ";
end for;

```

For the subgroups $\bar{H} = O_{10}^-(2)$, $SO_7(3)$, $Fi_{22}.2$ and $2 \times F_4(2)$, we use GAP to access the corresponding character tables and associated data on conjugacy classes. Here we work with the tuples a and b define above:

```

U:=["O10-(2).2","O7(3).2","Fi22.2","2xF4(2)"];
C1:=[20164, 793815, 36757748, 79942500];;
C2:=[730, 22464, 309960, 995085];;
C3:=[139232, 4524975, 355384576, 1061489520];;
C4:=[65287, 1219725, 41760576, 115887915];;
C:=[C1,C2,C3,C4];;
for i in [1..Size(U)] do
  t1:=CharacterTable(U[i]);;
  o1:=OrdersClassRepresentatives(t1);;
  s1:=SizesConjugacyClasses(t1);;

```

```

R:=PrimeDivisors(Size(t1));
B:=[];
for r in R do
  if r = 2 then
    D:=C[i];
    for j in [1..Size(D)] do
      if D[j]/a[j] > b[j] then
        Add(B,j);
      fi;
    od;
  else
    z:=0;
    for j in [1..Size(o1)] do
      if o1[j] = r then
        z:=z+s1[j];
      fi;
    od;
    for k in [1..Size(Q)] do
      if Q[k] = r then
        if z/a[k] > b[k] then
          Add(B,k);
        fi;
      fi;
    od;
  fi;
od;
Print(U[i],"\n",B,"\n","\n");
od;

```

For the parabolics, we switch back to MAGMA and we appeal to the information in [4, Tables 5 and 6].

```

h1:=2*2^24**SOMinus(8,2);
h2:=2*2^21**PSU(6,2)*2;
h3:=2*2^31**Alt(5)**PSL(3,2)*2;
h4:=2*2^29**PSL(3,4)*2*6;
C1:=[138296,4871775,355401728,1119260008,3315597312,79859548160];
C2:=[451640,16107103,927793152,2597999976,58617495552,51673825280];
C3:=[31800,595551,31129600,101543272,266338304,4697620480];
C4:=[66616,1443423,46202880,166458728,1763704832,4697620480];
H:=[h1,h2,h3,h4];
C:=[C1,C2,C3,C4];

```

```

for i in [1..4] do
  B:=[];
  h:=H[i];
  F:=PrimeFactors(h);
  D:=C[i];
  for j in [1..6] do
    if D[j]/a[j] gt b[j] then
      Append(~B,j);
    end if;
  end for;
  for j in [7..#P] do
    if (P[j] in F) and (h/a[j] gt b[j]) then
      Append(~B,j);
    end if;
  end for;
  i; B; " ";
end for;

```

```

t1:=CharacterTable(m[1]);
fus:=FusionConjugacyClasses(t1,t);
o1:=OrdersClassRepresentatives(t1);
A:=[];
for i in [1..Size(o1)] do

```

```

if o1[i] = 5 then
  Add(A,ClassNames(t){[fus[i]]});;
fi;
od;
A;

```

This returns [["5a"]], which tells us that L has a unique conjugacy class of elements of order 5, and they are contained in the G -class labelled 5A. Finally, we apply [7, Theorem 2], which tells us that if \bar{H} is a maximal parabolic subgroup and $x \in G$ has order 5, then $|x^G \cap H| = i_5(H) \leq |x^G|/192$ and it is easy to check that this bound is good enough.

```

s:=SizesConjugacyClasses(t);;
s1:=SizesConjugacyClasses(t1);;
A:=[];
B:=[];
c:=1/192;;
for i in [1..Size(o1)] do
  if o1[i] = 5 then
    Add(A,s1[i]);;
    Add(B,s[fus[i]]);;
  fi;
od;
(c*A[1])/B[1] < b[7];

```

□

4.3. Soluble maximal pairs. In this section, we present the computations used in the proof of [1, Proposition 4.20]. We begin by handling the groups with socle M_{11} , M_{12} or J_2 .

```

U:=["M11", "M12", "J2"];
for u in U do
  G:=AutomorphismGroupSimpleGroup(u);
  T:=Socle(G);
  if #G ne #T then
    a1,b1,c1:=SolMax(G,100);
    a2,b2,c2:=SolMax(T,100);
    A1:=[];
    A2:=[];
    for c in c1 do
      Append(~A1,[#c[1],#c[2]]);
    end for;
    for c in c2 do
      Append(~A2,[#c[1],#c[2]]);
    end for;
    u; [a1,b1]; A1; " "; [a2,b2]; A2; " "; " ";
  else
    a1,b1,c1:=SolMax(G,100);
    A1:=[];
    for c in c1 do
      Append(~A1,[#c[1],#c[2]]);
    end for;
    u; [a1,b1]; A1; " "; " ";
  end if;
end for;

```

Next we use GAP to handle the remaining groups with two or more conjugacy classes of soluble maximal subgroups, excluding the Monster. Note that $V[i]$ gives the labels of the soluble maximal subgroups of $G = U[i]$ (up to conjugacy, consistent with the labelling used by the `Maxes` function). So for example, $G = J_1$ has 5 classes of soluble maximal subgroups, labelled 2, 4, 5, 6 and 7.

```

U:=["J1", "J3", "J3.2", "He", "He.2", "McL.2", "O'N.2", "Co1", "HN", "HN.2", "Ly", "Th", "J4", "Fi24", "B"];;
V1:=[2,4,5,6,7];;
V2:=[7,9];;

```

```

V3:=[6,8,9];;
V4:=[8,11];;
V5:=[8,9,12];;
V6:=[9,10];;
V7:=[5,7,8];;
V8:=[15,21];;
V9:=[6,13];;
V10:=[7,12];;
V11:=[8,9];;
V12:=[6,7,9,11,15];;
V13:=[7,11,12,13];;
V14:=[20,21];;
V15:=[17,30];;
V:=[V1,V2,V3,V4,V5,V6,V7,V8,V9,V10,V11,V12,V13,V14,V15];;

for j in [1..Size(U)] do
  u:=U[j];;
  v:=V[j];;
  t:=CharacterTable(u);;
  s:=Maxes(t);;
  m:=[];
  for i in [1..Size(s)] do
    if i in v then
      Add(m,s[i]);;
    fi;
  od;
  k:=2;
  F:=[];
  for i in [1..Size(m)] do
    t1:=CharacterTable(m[i]);;
    Add(F,FusionConjugacyClasses(t1,t));;
  od;
  q:=QGAP(t,m,F,k);;
  if Size(q[1]) > 0 then
    c:=[];
    for i in [1..Size(q[1])] do
      Add(c,[s[v[q[1][i][1]]],s[v[q[1][i][2]]]]);;
    od;
  else
    c:=q[1];;
  fi;
  Print(u,"\\n",c,"\\n",q[2],"\\n","\\n");
od;

```

This returns an empty pair of tuples, with the exception of the following two cases:

```

J3.2
[ [ "2^(2+4):(S3xS3)", "2^(2+4):(S3xS3)" ] ]
[ ]

He.2
[ [ "2^(4+4).(S3xS3).2", "2^(4+4).(S3xS3).2" ] ]
[ ]

```

Finally, we handle the Monster $G = \mathbb{M}$:

```

t:=CharacterTable("M");;
s:=NamesOfFusionSources(t);;
v:=[33,38];;
m:=[];
for i in [1..Size(s)] do
  if i in v then
    Add(m,s[i]);;
  fi;
od;
k:=2;
F:=[];
for i in [1..Size(m)] do

```



```

t1:=CharacterTable(m[i]);;
Add(F,FusionConjugacyClasses(t1,t));;
od;
q:=QGAP(t,m,F,k);;
q[1];

```

This returns [] and thus $R_{\text{sol max}}(G) = 2$. This completes the proof of [1, Proposition 4.20].

4.4. Nilpotent-soluble pairs. In this final section, we consider the existence of non-regular pairs (H, K) , where H is nilpotent and K is soluble (see [1, Remark 4.22]).

First we handle the following groups:

$$M_{11}, M_{12.2}, M_{22.2}, M_{23}, M_{24}, J_2.2, \text{HS}.2$$

The following MAGMA code shows that such pairs only arise when $G = M_{22.2}$ or $J_2.2$.

```

U:=["M11","M12","M22","M23","M24","J2","HS"];
A:=[];

for u in U do
  G:=AutomorphismGroupSimpleGroup(u);
  T:=Socle(G);
  N1:=NilpotentSubgroups(G);
  S1:=SolubleSubgroups(G);
  N1:=[N1[i]'subgroup : i in [1..#N1] | #PrimeFactors(#N1[i]'subgroup) ge 2];
  S1:=[S1[i]'subgroup : i in [1..#S1] | #PrimeFactors(#S1[i]'subgroup) ge 2];
  P:=PrimeFactors(#G);
  for p in P do
    Append(~N1,SylowSubgroup(G,p));
    Append(~S1,SylowSubgroup(G,p));
  end for;
  D:=[#N1[i] : i in [1..#N1]];
  ParallelSort(~D, ~N1);
  N1:=Reverse(N1);
  D:=[#S1[i] : i in [1..#S1]];
  ParallelSort(~D, ~S1);
  S1:=Reverse(S1);
  u; [#N1,#S1];

  B:=[];
  for r in [1..#N1] do
    for s in [1..#S1] do
      H:=N1[r];
      K:=S1[s];
      w:=0;
      for i in [1..100] do
        x:=Random(G);
        if #(H meet K^x) eq 1 then
          w:=1;
          break i;
        end if;
      end for;
      if w eq 0 then
        Append(~B,[r,s]);
      end if;
    end for;
  end for;
  Append(~A,B);
  #B; " ";
end for;

```

And we can repeat for $G = M_{12}, M_{22}, J_2$ and HS in order to show that none of these simple groups admit a non-regular nilpotent-soluble pair:

```

U:=["M12","M22","J2","HS"];
A:=[];

for u in U do

```

```

G:=AutomorphismGroupSimpleGroup(u);
G:=Socle(G);

N1:=NilpotentSubgroups(G);
S1:=SolubleSubgroups(G);
N1:=[N1[i]'subgroup : i in [1..#N1] | #PrimeFactors(#N1[i]'subgroup) ge 2];
S1:=[S1[i]'subgroup : i in [1..#S1] | #PrimeFactors(#S1[i]'subgroup) ge 2];
P:=PrimeFactors(#G);
for p in P do
  Append(~N1,SylowSubgroup(G,p));
  Append(~S1,SylowSubgroup(G,p));
end for;
D:=[#N1[i] : i in [1..#N1]];
ParallelSort(~D, ~N1);
N1:=Reverse(N1);
D:=[#S1[i] : i in [1..#S1]];
ParallelSort(~D, ~S1);
S1:=Reverse(S1);
u; [#N1,#S1];

B:=[];
for r in [1..#N1] do
  for s in [1..#S1] do
    H:=N1[r];
    K:=S1[s];
    w:=0;
    for i in [1..100] do
      x:=Random(G);
      if #(H meet K^x) eq 1 then
        w:=1;
        break i;
      end if;
    end for;
    if w eq 0 then
      Append(~B,[r,s]);
    end if;
  end for;
end for;
Append(~A,B);
#B; " ";
end for;

```

We claim that if $G \in \{Fi_{22}, Fi_{22}.2, Co_2, Fi_{23}\}$, then every nilpotent-soluble pair is regular. To do this, we proceed as above, but we replace `SolubleSubgroups` by our function `MaxSolubleSubgps`. Similarly, we replace `NilpotentSubgroups` by the following function, which returns a set $\mathcal{N}_d(H)$ of subgroups of H , which is the exact analogue of $\mathcal{S}_d(H)$ for nilpotent subgroups (see [1, Definition 4.14]).

```

MaxNilpSubgps := function(G,H,d)

a:=1;
S:=[];
NS:=[];
if IsNilpotent(H) then
  Append(~S,H);
else
  Append(~NS,H);
end if;

while (#NS ge 1) and (a le d) do
  NS1:=[];
  for H in NS do
    M1:=MaximalSubgroups(H);
    M1:=[M1[i]'subgroup : i in [1..#M1]];
    for i in [1..#M1] do
      b:=exists(K){K : K in S | IsConjugate(G,K,M1[i])};
      if IsNilpotent(M1[i]) and (b eq false) then

```

```

        Append(~S,M1[i]);
    else
        Append(~NS1,M1[i]);
    end if;
end for;
end for;
NS:=NS1;
a:=a+1;
end while;

if #NS ge 1 then
    for H in NS do
        b:=exists(K){K : K in S | IsConjugate(G,K,H)};
        if b eq false then
            Append(~S,H);
        end if;
    end for;
end if;

S:=[S[i] : i in [1..#S] | #PrimeFactors(#S[i]) ge 2];
P:=PrimeFactors(#H);
for p in P do
    Append(~S,SylowSubgroup(H,p));
end for;

D:=[#S[i] : i in [1..#S]];
ParallelSort(~D, ~S);
S:=Reverse(S);
return S;

end function;

```

Proposition 4.14. *If $G \in \{\text{Fi}_{22}, \text{Fi}_{22}.2, \text{Co}_2, \text{Fi}_{23}\}$, then every nilpotent-soluble pair is regular.*

Proof. For $G = \text{Fi}_{22}$, we proceed as follows, working with the subsets $\mathcal{N}_5(G)$ and $\mathcal{S}_5(G)$ obtained via `MaxNilpSubgps` and `MaxSolubleSubgps`, respectively.

```

U:=["Fi22"];
A:=[];

for u in U do
    G:=AutomorphismGroupSimpleGroup(u);
    G:=Socle(G);
    N1:=MaxNilpSubgps(G,G,5);
    S1:=MaxSolubleSubgps(G,G,5);
    B:=[];
    for r in [1..#N1] do
        for s in [1..#S1] do
            H:=N1[r];
            K:=S1[s];
            w:=0;
            for i in [1..100] do
                x:=Random(G);
                if #(H meet K^x) eq 1 then
                    w:=1;
                    break i;
                end if;
            end for;
            if w eq 0 then
                Append(~B,[r,s]);
            end if;
        end for;
    end for;
    Append(~A,B);
    #B; " ";
end for;

```

B;

This returns [] and we conclude that every nilpotent-soluble pair is regular. We can repeat the calculation for $G = \text{Fi}_{22}.2 = T.2$, noting that we only need to show that every nilpotent-soluble pair (H, K) with $H, K \not\leq T$ is regular.

If we repeat for $G = \text{Co}_2$, then the above procedure returns two inconclusive pairs (H, K) , where K is soluble and H is non-nilpotent. To handle these cases, we apply the function `MaxNilpSubgps` to construct the set $\mathcal{N}_3(H)$ and we then use random search to check that each pair (J, K) with $J \in \mathcal{N}_3(H)$ is regular.

Finally, the group $G = \text{Fi}_{23}$ can be handled in a similar fashion. We begin by constructing the sets $\mathcal{N}_4(G)$ and $\mathcal{S}_4(G)$, and we find that there are 104 inconclusive pairs (H, K) with $H \in \mathcal{N}_4(G)$ and $K \in \mathcal{S}_4(G)$. For each of these pairs, we construct $\mathcal{N}_3(H)$ and $\mathcal{S}_3(K)$, and then random search shows that each (J, L) with $J \in \mathcal{N}_3(H)$ and $L \in \mathcal{S}_3(K)$ is regular. In this way, we conclude that every nilpotent-soluble pair for Fi_{23} is regular. \square

REFERENCES

- [1] M. Anagnostopoulou-Merkouri and T.C. Burnes, *On the regularity number of a finite group and other base-related invariants*, preprint.
- [2] W. Bosma, J. Cannon and C. Playoust, *The MAGMA algebra system I: The user language*, J. Symb. Comput. **24** (1997), 235–265.
- [3] T. Breuer, *The GAP Character Table Library, Version 1.3.3*, GAP package, <http://www.math.rwth-aachen.de/~Thomas.Breuer/ctbllib>, 2022.
- [4] T.C. Burnes, *On soluble subgroups of sporadic groups*, Israel J. Math. **254** (2023), 313–340.
- [5] T.C. Burnes, E.A. O’Brien and R.A. Wilson, *Base sizes for sporadic simple groups*, Israel J. Math. **177** (2010), 307–333.
- [6] The GAP Group, *GAP – Groups, Algorithms, and Programming*, Version 4.11.1, 2021 (<http://www.gap-system.org>).
- [7] R. Lawther, M.W. Liebeck and G.M. Seitz, *Fixed point ratios in actions of finite exceptional groups of Lie type*, Pacific J. Math. **205** (2002), 393–464.
- [8] S.A. Linton and R.A. Wilson, *The maximal subgroups of the Fischer groups Fi_{24} and Fi'_{24}* , Proc. London Math. Soc. **63** (1991), 113–164.
- [9] J. Müller, M. Neunhöffer, F. Noeske and M. Horn, *ORB – methods to enumerate orbits*, Version 4.8.3, 2019. <https://gap-packages.github.io/orb>
- [10] R.A. Wilson et al., *A World-Wide-Web Atlas of finite group representations*, <http://brauer.maths.qmul.ac.uk/Atlas/v3/>

M. ANAGNOSTOPOULOU-MERKOURI, SCHOOL OF MATHEMATICS, UNIVERSITY OF BRISTOL, BRISTOL BS8 1UG, UK

Email address: marina.anagnostopoulou-merkouri@bristol.ac.uk

T.C. BURNES, SCHOOL OF MATHEMATICS, UNIVERSITY OF BRISTOL, BRISTOL BS8 1UG, UK

Email address: t.burnes@bristol.ac.uk